



[返回总目录](#)

目 录

第 24 章	SAS 系统内七种变异数分析程序概述.....	4
24.1	七种变异数分析的程序.....	4
24.2	平衡的实验设计.....	4
24.3	一般线性模型.....	5
第 25 章	比较两组平均数的 t 检定：统计程序 PROCTTEST	7
25.1	PROC TTEST 程序概述	7
25.2	如何撰写 PROC TTEST 程序	7
25.3	输出文件概述.....	8
25.4	范 例.....	9
第 26 章	平衡实验设计的变异数分析：统计程序 PROCANOVA.....	11
26.1	PROC ANOVA 程序概述	11
26.2	名 词 解 释.....	11
26.3	各种统计模型.....	11
26.4	如何撰写 PROC ANOVA 程序	13
26.5	范 例.....	21
第 27 章	变异数成份的分解：统计程序 PROCVARCOMP	27
27.1	PROC VARCOMP 程序概述	27
27.2	如何撰写 PROC VARCOMP 程序	28
27.3	范 例.....	29
第 28 章	混合式模型的变异数分析：统计程序 PROC MIXED	33
28.1	PROC MIXED 程序概述	33
28.2	名 词 解 释.....	33
28.3	MIXED 程序基本功能的示范.....	34
28.4	如何撰写 PROC MIXED 程序	38
28.5	范 例.....	51
28.6	注 意 事 项.....	68
第 29 章	变异数分析的实验设计：统计程序 PROC PLAN.....	70
29.1	PROC PLAN 程序的简介.....	70
29.2	如何撰写 PROC PLAN 程序.....	71
29.3	范 例.....	75
29.4	注 意 事 项.....	80
第 30 章	无参数的一因子变异数分析：统计程序 PROCNPAR1WAY	81
30.1	PROC NPAR1WAY 程序概述	81
30.2	如何撰写 PROC NPAR1WAY 程序	81

30.3 范 例.....	83
30.4 注 意 事 项.....	91

禁书网电子出版社版权所有

第五部分

变 异 数 分 析

第 24 章 SAS 系统内七种变异数分析程序概述

24.1 七种变异数分析的程序

本节提纲契领地介绍 SAS 所提供的七个变异数 (analysis of variance) 分析的程序及一些有关的统计观念。

这七个程序简介如下：

TTEST	以 t 检定比较两组观察体的平均数。
ANOVA	适用于平衡的实验设计。
GLM	功能甚广，可执行变异数分析、回归分析、共变量分析以及多变量变异数分析 (归入第六部分第 31 章)。
VARCOMP	推算各类型变异数的值。
MIXED	适用于固定与随机效果的混合式模型分析。
PLAN	为实验计划产生随机的排列组合。
NPARIWAY	适用于无参数分析中单因变量之变异数分析。

这七个程序中，以 GLM 与 MIXED 两程序功能最广泛，其他则只适用于特殊的情况。读者必须依照实验设计选用合适的 SAS 程序。变异数分析的用途在于解释我们观察所得的数据。一般而言，这些数据是在不同的实验情况下收集来的。这些不同的情况会造成数据间的异同，此称实验效果 (Treatment Effect)。但另有一部分的异同是与实验情况无关的，此称随机误差 (Random Error)。所有变异数分析的精髓不外乎是选定合适的统计模型，利用正确的 SAS 程序来推算出实验效果和随机误差的平均方 (Mean Square)。若以 MS(A) 表实验效果的平均方；以 MS(E) 表随机误差的平均方，则其比例会导出一个 F 分配：

$$F = \frac{MS(A)}{MS(E)}$$

F 的自由度随不同的实验设计而定。

若 MS(A) 远超过 MS(E) (即 F 值远大于 1)，则我们说实验效果显著。反之，我们说实验效果不显著。这个统计的理论是费契尔 (Fisher, 1925) 所建立的，有关这个理论最早的教科书是沙菲所著 (Scheffe, 1959)。

24.2 平衡的实验设计

平衡的实验设计是指相等人数的实验；也就是说每一组 (或每一细格) 里的观察体个数相同。若研究者的实验设计是平衡的，则你可以用较简单的 ANOVA 程序计算出变异数分析中所有必需的统计值，而不必用到较费时费力的 GLM 程序。若把 ANOVA 程序

用在不平衡的实验设计计划，则结果会有误差，甚至可能导出负的平均方值。

24.3 一般线性模型

如果你的实验设计是不平衡的，则你不可用 ANOVA 程序，你极可能会用到 PROC GLM (一般线性模型程序)，或 PROC MIXED (混合式的线性模型)。

在统计领域中的线性模型与解析几何中的线性函数或线性图有异同之处。相同之处是它们都用“线性”这个名词来表示一个“一次”的或线性的函数关系。相异之处是统计中所指的线性关系是指因变量与参数之间的线性关系，而非因变量与自变量之间的线性关系。

若以 X 代表自变量， Y 代表因变量， β 代表参数， E 代表随机误差，则下列三式都称为线性模型：

$$\begin{aligned}\sqrt{Y_i} &= \beta_1 X_1 + \beta_2 X_2 + E_i \\ Y_i &= \beta_1 X_1^2 + \beta_2 X_2 + E_i \\ Y_i &= \beta_1 \log(X_1) + \beta_2 \sqrt{X_2} + E_i\end{aligned}$$

但下列三式则称为非线性模型：

$$\begin{aligned}Y_i &= e^{(\beta_1 X_1 + \beta_2 X_2)} + E_i \\ Y_i &= \sqrt{\beta_1 X_1 + \beta_2 X_2} + E_i; \text{或} \\ Y_i &= \frac{\beta_1}{\beta_1 - \beta_2} (e^{-\beta_1 X_1} - e^{-\beta_2 X_2}) + E_i\end{aligned}$$

许多的非线性模型可被转换成线性模型，否则它们无法用变异数分析法来处理。

■ 线性假设

线性假设是指参数间的线性组合而言，下面列举几种常见的表示法：

- H: $\beta_1 = \beta_2 = \dots = 0$; 或
- H: $L_1 \beta_1 + L_2 \beta_2 + L_3 \beta_3 + \dots + L_k \beta_k = 0$; 或
- H: $L\beta = 0$ (此处， L 是行向量， β 是列向量)

■ 随机效果

随机效果一般是由随机因子导出。在农业经济研究上，农作物的产地（如：畦田或耕地等）通常被视为一个随机因子 (Random Factor)。在教育界，一个班级或学校或一群学生则被看作是一个随机因子。随机因子的效果称为随机效果。关于这些随机因子的变异数分析，SAS 预备有 VARCOMP 与 NESTED 两程序。另一程序 GLM 则只印出随机因子的平均方值，但不推算各类型变异数的预计值。

■ 平均数的比较

如果数据中含多个平均数，而且读者已决定要比较哪两个平均数时，可用 GLM 程

序中的 CONTRAST 指令来检验这两个平均数之间是否有显著的差异。请读者注意，若你反复地使用 CONTRAST 指令来测许多对平均数，则你的分析结果将失去真实性。这是因为这种分析法可能让你犯了统计上所谓的第一类型错误。详情请见第 31 章 PROC GLM 的平均数比较部分。

■无参数的变异数分析

无参数的变异数分析并不要求数据符合常态分配的假设，因此数据不能以一般参数分析的统计方法处理。在这种情况下，可用 NPARIWAY 程序将数据转换成名次排列(如：第一、第二等等)，然后进行无参数的单因变量变异数分析。二元或二元以上无参数的变异数分析与 NPARIWAY 相似。读者可先用 PROC RANK 把数据转换成名次，然后再用一般的参数分析程序处理，如：TEST, ANOVA 或 NESTED 等。

第 25 章 比较两组平均数的 t 检定：统计程序 PROC TTEST

25.1 PROC TTEST 程序概述

TTEST 程序旨在对 SAS 文件中的两个平均数执行 t 检定。这个 t 检定是单元变异数分析的特殊例子，它的虚无假设是这两个平均数相等。

t 检定有一个重要的假设，即：两组观察体所代表的母群，其变异数必须相同。这个假设由 F 检定来鉴别 (Steel and Terrie, 1980)。若此假设成立，则 SAS 可进一步算出 t 值与其统计的显著程度。若此假设不成立，则 SAS 会算出 t 的近似值与其近似的自由度。自由度的近似值估计系采沙特斯威氏 (Satterthwaite, 1946) 的方法。同时读者可要求根据 Cochran 与 Cox (1957) 计算 t 检定之近似值的显著度。

若读者有意进行一组平均数的 t 检定或配对组平均数比较的 t 检定，则应用 PROC MEANS 来进行 (见本章例二的示范)，TTEST 程序只适用于两组独立的样本。

25.2 如何撰写 PROC TTEST 程序

PROC TTEST 含四道指令，它们的格式如下：

```
PROC TTEST 选项串;  
  CLASS 变量名称;  
  VAR 变量名称串;  
  BY 变量名称串;
```

上述四道指令不能重复出现。只有 PROC TTEST 与 CLASS 两指令是必须的，不可省略。PROC TTEST 后的指令出现次序可以随意安排。

指令 #1 PROC TTEST 选项串:

下面这两个选项可出现在 PROC TTEST 指令中：

(1) DATA=输入文件名称

指明对那一个文件执行 t 检定。若省略此选项，则 SAS 会自动找出在此程序之前最后形成的 SAS 文件，对它执行 t 检定。

(2) COCHRAN

当两组数据的变异数据不相当时，这个选项可以正确地计算出 t 检定之近似值的统计显著度，其理论基础是 Cochran 与 Cox 1957 年的著作。

指令 #2 CLASS 变量名称:

此变量旨在识别观察体所属的组别, 因此变量只可有两个不同的值 (如: 男, 女或 1, 0 等)。如果此变量的值是英文字母, 如: MALE 或 FEMALE, 则名字的长度不应超过十六个字母, 否则警告信息会出现。

指令 #3 VAR 变量名称串:

指明对那些因变量的平均数执行 t 检定。若省略此指令, 则 SAS 会视输入文件内所有数值变量 (除 CLASS 指令里提到的变量外) 为因变量。然后针对每一数值变量执行 t 检定。

指令 #4 BY 变量名称串:

SAS 依据此指令所列举的变量将文件分成几个小的文件, 然后对每一个小的文件分别执行 t 检定。当读者选用此指令时, 文件内的数据必须先按照 BY 变量串的值做由小到大的重新排列。这个步骤可藉 PROC SORT 达成。

25.3 输出文件概述

针对每一个参与 t 检定的因变量, TTEST 程序会印出下列的统计值:

1. 该因变量的名称
2. 组别名称
3. 有效观察体个数 (N)
4. 平均数 (MEAN)
5. 标准差 (STD DEV)
6. 标准误差 (STD ERROR)
7. 最小值 (MINIMUM)
8. 最大值 (MAXIMUM)

当等值变异数的假设不成立时, 印出以下的统计值:

9. t 检定的近似值 (T)
10. 近似的自由度 (DF)
11. 双尾检定的显著程度 (PROB>T)

当等值变异数的假设成立时, SAS 改印:

12. t 检定的正确值 (T)
13. 正确的自由度 (DF)
14. 双尾检定的显著程度 (PROB>T, 与上述 (11) 同)

等值变异数假设的检验结果是以下列的统计值来表示:

15. 鉴别等值变异数假设的 F' 值 (F')
16. F' 检定的自由度 (DF)
17. 大于现有 F' 值的单尾检定的显著程度 (PROB>F')

25.4 范 例

例一：以 PROC TTEST 比较两个独立样本的平均数

本例采用一班学生的体育成绩来示范 PROC TTEST。这一班学生有男有女，我们想知道到底男女学生在网球的球技上是否有显著的不同。虚无假设则是男女球技相同。

程 序

```
DATA SCORES;
    INPUT SEX $ SCORE @@;
    CARDS;
    F 75 F 76 F 80 F 77 F 80 F 77 F 73
    M 82 M 80 M 85 M 85 M 78 M 87 M 82
    ;
PROC TTEST;
    CLASS SEX;
    VAR SCORE;
    TITLE 'GOLF SCORES';
RUN;
```

结 果

首先，看 F' 对变异数等值的假设检定： $F'=1.53$ ，未达显著程度。因此，接下来我们可以接受 Variances=Equal 的 t 检定值： $T=-3.8288$ (自由度=12.0)，达 0.0024 的显著程度。所以，结论是男女在网球的球技上有高下之分。

报表 25.1 以 PROC TTEST 比较两个独立样本的平均数

GOLF SCORES						
TTEST PROCEDURE						
Variable: SCORE						
SEX	N	Mean	Std Dev	Std Error	Minimum	Maximum
F	7	76.85714286	2.54483604	0.96185761	73.00000000	80.00000000
M	7	82.71428571	3.14718317	1.18952343	78.00000000	87.00000000
Variances	T	DF	Prob> T			
Unequal	-3.8288	11.5	0.0026			
Equal	-3.8288		12.0	0.0024		
For H0:Variances are equal, F'=1.53				DF= (6, 6)	Prob>F'=0.6189	

例二：以 PROC MEANS 比较配对组的平均数

当两组数据之间有相关时 (如：夫妻文件的数据，或同一班学生前后两次考试的成绩)，则读者应用 PROC MEANS (而非 PROC TTEST) 来比较这两组之间平均数的差异。本例另创一个新变量 (DIFF) 来代表两次考试的平均数差。PROC MEANS 的两个选项 T

与 PRT 引导 SAS 进行配对组的 t 检定而且计算其统计显著度。

程 序

```
DATA A;
    INPUT ID PRETEST POSTTEST;
    DIFF=POSTTEST-PRETEST;
    CARD;
1 80 82
2 73 71
3 70 95
4 60 69
5 88 100
6 84 71
7 65 75
8 37 60
9 91 95
10 98 99
11 52 65
12 78 83
13 40 60
14 79 86
15 59 62
;
PROC MEANS MEAN STDERR T PRT;
    VAR DIFF;
    TITLE 'PAIRED-COMPARISONS T TEST';
RUN;
```

结 果

分析结果显示后测比前测平均高出 7.93 分。这个差异经 t 检定检验后 (T=3.09), 证明达 0.0079 的显著度。

报表 25.2 以 PROC MEANS 比较配对组的平均数				
PAIRED-COMPARISONS T TEST				
Analysis Variable : DIFF				
NObs	Mean	StdError	T	Prob> T
15	7.9333333	2.5643465	3.0937057	0.0079

第 26 章 平衡实验设计的变异数分析：统计程序 PROC ANOVA

26.1 PROC ANOVA 程序概述

ANOVA 程序主要是对平衡实验设计的数据执行变异数分析；但也可以处理拉丁方格实验设计，完全的镶嵌设计 (Completely Nested Design)，细格之间人数成比例的实验设计等。所谓的平衡实验设计是指组间 (或细格间) 人数相等的实验设计。不平衡的实验设计则不可用 ANOVA 程序来处理，要用 GLM 程序 (见第 31 章)。

26.2 名词解释

■ 自变量与因变量

自变量又称独立变量、定性变量 (Qualitative Variable)、分类变量 (Classification Variable)、或是类别变量 (Categorical Variable)；其数值多半是不连续的。反之，因变量又称反应变量 (Response Variable)；其数值则是连续的。

■ 实验效果

变异数分析的目的在于找出自变量与因变量之间的线性关系，或说自变量对因变量产生的实验效果。这种实验效果可粗分为三种，即：主效果、交互效果、与镶嵌效果。主效果以自变量的英文字母代表，如 A, B 等。交互效果以星号联接的自变量表示，如：A*B。镶嵌效果以小括号表示，如：A(B) 表示 A 效果是镶嵌在 B 效果内。

26.3 各种统计模型

上述的三种实验效果分别与变异数分析法中不同的统计模型相对应。现将这三种统计的模型分述如下：

■ 主效果的统计模型

假设有一个平衡的实验设计，含三个自变量 (分别以 A, B, C 表示)，其因变量以 Y 表示，则此三因子主效果变异数分析可以下面的程序来执行：

```
PROC ANOVA;  
  CLASS A B C;  
  MODEL Y = A B C;
```

■交互效果的统计模型

这种模型适用于含两个或两个以上自变量的实验设计。若以上述的三因子设计为例，其对应的主效果及交互效果可用下列的程序来计算：

```
PROC ANOVA;
  CLASS A B C;
  MODEL Y = A B C A*B B*C A*C A*B*C;
```

当实验设计含多个自变量时，交互效果会变得繁杂。此时可用竖号 | 来简化。比方说，上例的 MODEL 指令可利用竖号简化如下：

MODEL Y = A|B|C; 等于
MODEL Y = A B C A*B B*C A*C A*B*C;

另外举几个使用竖号的例子如下：

A|C(B) 等于 A C(B) A*C(B)
A(B)|C(B) 等于 A(B) C(B) A*C(B)
A(B)|B(D E) 等于 A(B) B(D E)
A|B(A)|C 等于 A C B(A) A*C B*C(A)

其他有关竖号的使用规则，请参考第 16 章 PROC CATMOD 第 16.3 节的内容。

此外，@ 的符号表交互作用的最高元次，因此

A|B|C@2 等于 A B C A*B A*C B*C
A|B(A)|C@2 等于 A B(A) C A*C
A|B|C|D@2 等于 A B A*B C A*C B*C D A*D B*D C*D

■镶嵌效果的统计模型

以上述三因子变量的实验设计为例，我们现在假设自变量 C 是镶嵌在另外两个自变量 A 与 B 内，则可用下列的 SAS 程序来执行变异数分析：

```
PROC ANOVA;
  CLASS A B C;
  MODEL Y = A B C(A B);
```

由上式可知，C 是写在 A 与 B 的小括号外面。这种写法表示 C 效果是镶嵌在 A 与 B 的交互作用内。

■其他注意事项

假如有一个实验设计同时包含了交互与镶嵌效果，则读者可同时使用 * 与 () 来表示，如：

```
PROC ANOVA;
  CLASS A B C;
  MODEL Y = A B(A) C(A) B*C(A);
```

B*C(A) 表示 B 与 C 的交互效果是镶嵌在 A 的主效果里。

值得读者注意的是：当 MODEL 指令中省略一些效果时，这些被省略的效果会自动与细格内的误差（即 Within-Cell Error）协调。根据这个原则，下面两道指令所表示的统计模型是完全相同的：

```
CLASS A B;
MODEL Y = A B(A);
```

等于

```
CLASS A B;
MODEL Y = A A*B;
```

上面两段指令皆省略了 B 的主效果，所以它们的细格内的误差值均相等（因都包含了 B 的主效果）。

26.4 如何撰写 PROC ANOVA 程序

PROC ANOVA 含十道指令，它们的格式如下：

PROC ANOVA	选项串；
CLASS	变量名称串；
MODEL	因变量名称串=实验效果串 (@)/ 选项串；
MEANS	实验效果串 / 选项串；
ABSORB	变量名称串；
FREQ	变量名称；
TEST	H= 效果名称 E= 效果名称；
MANOVA	H= 效果名称 E= 效果名称 M= 变量的转换式 PREFIX= 新变量的名称代号 MNames= 新变量的名称串 / 选项串；
REPEATED	重复变量的名称 组数据(组名) 变量的转换 / 选项串；
BY	变量名称串；

CLASS 指令必须出现在 MODEL 指令之前。如果选用 TEST, MANOVA 指令，则它们必须出现在 MODEL 指令之后。MEANS, TEST 及 MANOVA 等指令可重复使用，其他指令则只能出现一次。

指令 #1 PROC ANOVA 选项串：

下面四个选项可放在 PROC ANOVA 指令之后：

(1) A DATA= 输入文件名称

指明对那一个 SAS 文件执行 ANOVA 分析。若省略此选项，则 SAS 会自动找出在此程序之前最后形成的 SAS 文件，对它执行 ANOVA 分析。

(2) MANOVA

要求 PROC ANOVA 将含一个或一个以上因变量遗漏数据的观察体剔除。当读者以交互式 (Interactive Mode) 方式进行多变量的变异数分析时, 最好界定此选项。

(3) MULTIPASS

要求 PROC ANOVA 在必要情况下重读输入文件内的数据。由于这个选项会占用极多的记忆体, 同时耗时很多, 除非必要, 读者可以省略此选项。

(4) OUTSTAT=(含分析结果的) 输出文件名称

这个选项会界定一个含分析结果的输出文件。此输出文件将含离差平方和 (SS)、F 检定值, 以及各实验效果的显著程度。若读者同时界定 MANOVA 指令中的 CANONICAL 选项但未界定 M= 的选项, 则典型相关分析的结果也会纳入此输出文件内。

指令 #2 CLASS 变量名称串;

此指令指明上述的文件中哪些变量是自变量。自变量可以是数值的 (如: 1 代表男, 2 代表女) 或文字的 (如: MALE 代表男, FEMALE 代表女)。若是文字变量, 则其长度不可超过十六个字母。

指令 #3 MODEL 因变量名称串=实验效果串 (@) / 选项串;

有关删除号 (/) 前 因变量= 实验效果 的部分, 读者必须自行决定适合的统计模型, 然后根据本章第 26.3 节所介绍的原则写出。@ 符号可以限制交互作用的最高元次。例如: A|B|C|D@2 则表示只需计算两两自变量相乘的交互作用即可。至于删除号 (/) 之后的选项有两个:

(1) A NOUNI

抑制单变量变异数分析结果的印出。这个选项适用于多变量的变异数分析或重复观察实验中有关重复变量的分析报表。

(2) INTERCEPT (或 INT)

要求 SAS 把线性模型内的截距 (亦即数据的总平均数) 当成一个参数, 同时对这个截距作是否为零的假设统计检定。

指令 #4 MEANS 实验效果串 / 选项串;

此指令的前半部 (删除号之前) 是用来要求 ANOVA 程序算出某些自变量 (和其交互作用或镶嵌作用) 中各组 (或细格) 的平均数。比方说我们可用下列的 SAS 程序算出文件中男人、女人、黑人、白人、男黑人、男白人、女黑人及女白人在因变量年薪 (SALARY) 上的平均数:

```
PROC ANOVA;
  CLASS SEX RACE;
  MODEL SALARY = SEX RACE;
  MEANS SEX RACE SEX*RACE;
```

删除号 (/) 之后可用的选项有二十四个。前十七个选项分别对 MEANS 指令中所列的主效果平均数执行不同的显著性检定。以上例而言, MEANS 指令会比较男与女, 黑人与白人之间的年薪差异。后七个选项则与统计检定的各项事宜有关。

(1) A BON

执行显著性 t 检定。其理论基础是班弗尼氏的不等律 (Bonferroni Inequality)。

(2) DUNCAN

执行唐肯氏多范围检定 (Duncan's Multiple-Range Test)。

(3) DUNNETT (控制组组别)

这个选项界定唐那氏的两组平均数之双尾检定。唐那氏 (Dunnett) 的检定依据 t 分配而且必须是实验组与控制组平均数的比较, 因此, 括号内必须指明控制组的组别, 请看下面的程序:

```
MEANS A/DUNNETT ('CONTROL');
```

根据这个指令的语法, A 效果的第 CONTROL 组就是控制组。若控制组的组别是以数字来表示的 (如 2), 则不必再加单引号, 如:

```
MEANS A/ DUNNETT (2);
```

这个选项的控制组一般是设定在第一个 (内设值)。若控制组不只一组时, 读者可同时在括号内提及, 如:

```
MEANS A B C/DUNNETT ('FIRST' 'SECOND' 'THIRD');
```

根据上述指令的语法, A 效果的控制组是第 FIRST 组, B 效果的控制组是第 SECOND 组, C 效果则是第 THIRD 组。

(4) DUNNETTL (控制组组名)

这个选项界定唐那氏的两组平均数之单尾检定, 而且预期的差异必须是负值 (亦即实验组的平均数小于控制组的平均数), 因此临界值订在 t 分配的下端。

有关控制组的内设值以及撰写语法, 请参见上面 (3) DUNNETT 的说明。

(5) A DUNNETTU (控制组组名)

这个选项界定唐那氏的两组平均数之单尾检定, 而且预期的差异必须是正值 (亦即实验组的平均数大于控制组的平均数), 因此临界值订在 t 分配的上端。

有关控制组的内设值以及撰写语法, 请参见上面 (3) DUNNETT 的说明。

(6) GABRIEL

执行贵博氏的多重比较 (Gabriel's Multiple-Comparison Procedure)。

(7) REGWF

执行 Ryan-Einot-Gabriel-Welsch 的 F 检定。

(8) REGWQ

执行 Ryan-Einot-Gabriel-Welsch 的 t 检定。

(9) SCHEFFE

执行沙菲氏 (Scheffe) 的多重比较检定。

(10) SIDAK

执行 Sidak 的两组平均数的 t 检定。

(11) SMM [或 (12) GT2]

执行 Sidak 的独立样本 t 检定。当两组人数不等时, 此法也就是哈氏 (Hochberg) 的 GT2 法。

(13) SNK

执行纽曼-库尔 (Newman-Keuls) 的两组平均数差的 t 检定。

(14) T [或 (15) LSD]

执行配对组 t 检定。因为 ANOVA 所处理的是平衡的设计, 故其结果与费契尔的最小显著差 (LSD) 的检验结果相同。

(16) TUKEY

执行土其氏 (Tukey) 的 HSD 检定。

(17) WALLER

执行 Waller-Duncan K-ratio 的 t 检定。

(18) ALPHA= P

界定统计检验的显著程度。内设值是 .05。当上面选项与选项 (2) DUNCAN 并用时, ALPHA 的值必须是 .10, .05 或 .01 三者之一。与上面其他检定选项并用时, ALPHA 可以是 0.0001 与 0.9999 间任何的值。

(19) LINES

将读者选用的显著性检定的分析结果 (即各平均数) 作由大到小的排列。若某一对平均数之间无显著的差异, 则 SAS 将它们印在同一行上, 并以虚线将它们与其它有显著差异的平均数分开。当读者选用 DUNCAN, REGWF, REGWQ, SNK 或 WALLER 等检定时, 此选项会自动被包括在内, 否则读者必须另外附加。此选项最适用于平衡的实验设计以及组数少于 24 的平均数比较。若细格内的人数不等, 则 ANOVA 程序计算各细格人数的调和平均数 (Harmonic Mean), 并用此数来比较主效果的平均数差异。当细格间人数差异太大时, 则某些比较的结果会不够严谨。

此选项不可与 (3) DUNNETT、(4) DUNNETTL 或 (5) DUNNETTU 联用。

(20) CLDIFF

将 BON, GABRIEL, SCHEFFE, SIDAK, SMM, GT2, T, LSD 或 TUKEY 显著性检定的结果用信赖区间的方式表示。当实验设计是一个不平衡的设计时, CLDIFF 选项会自动被包括在内。当读者选用 DUNCAN, REGWF, REGWQ, SNK 或 WALLER 时, 则须另外附加。

(21) CLM

将 MEANS 指令中所提到的效果的各组平均数以信赖区间的方式表示。此选项必须与 BON, GABRIEL, SCHEFFE, SIDAK, SMM, T, 以及 LSD 等联用。

(22) NOSORT

与上述 CLDIFF 或 CLM 选项合用, 抑止平均数按大小重新作排列。

(23) E=效果名称

此选项界定上述各显著性检定的分母。若省略此选项，则实验设计的误差 (或余差) 的平均方 (MS Residual) 就自动成为分母。

(24) KRATIO=正整数

与 WALLER 选项联用。这个比例 (第一类型错误 / 第二类型错误) 的值若订为 50, 100 或 500, 则大约与 ALPHA 值 .10, .05, .01 相对应。这个选项的内设值是 100。

指令 #5 ABSORB 变量名称串:

此指令的用途旨在节省计算时间与电脑的记忆储存量。详情请见第 31 章 (PROC GLM) 的 31.8 节 (ABSORB 指令及其使用方法)。

请读者注意：当选用此指令时，文件中的数据必须依 ABSORB 指令中的变量做由小到大的排列；而且这些变量不可出现在 CLASS 或 MODEL 指令内，否则会产生错误的平方总和。

指令 #6 FREQ 变量名称:

此变量的值就是文件中各观察体重复出现的次数。

指令 #7 TEST H=效果名称 E=效果名称:

一般而言，SAS 的变异数分析自动采用误差 (或余差) 的平均方 (MS Residual) 作为 F 检定的分母。读者可藉此指令自订 F 检定的分子、分母以进行不同的 F 检定；其中 H= 分子，而 E= 分母。请看下例：

```
PROC ANOVA;
  CLASS A B C;
  MODEL Y = A|B(A)|C;
  TEST H = A E = B(A);           [所以, F = A/B(A)]
  TEST H = C A*C E = B*C(A);    [所以, F = C/B*C(A), F = A*C/B*C(A)]
```

指令 #8 MANOVA H=效果名称 E=效果名称

M=变量的转换式

PREFIX=新变量的名称代号

MNAMES=新变量的名称串 / 选项串;

当 MODEL 指令中含一个以上的因变量时，读者可利用 MANOVA 指令，要求执行多变量变异数分析 (Multivariate Analysis of Variance)。

MANOVA 指令的写法示范如下：

```
PROC ANOVA;
  CLASS A B;
  MODEL Y1-Y5 = A B(A);
  MANOVA H = A E = B(A) / PRINTH PRINTE;
  MANOVA H = B(A) / PRINTE;
```

```
MANOVA H = A E = B(A) M = Y1-Y2, Y2-Y3, Y3-Y4, Y4-Y5
PREFIX =DIFF;
```

上列的程序里，由于在 MODEL 指令中有五个因变量 (Y1, Y2, Y3, Y4, Y5)，故可利用 MANOVA 指令要求执行多变量变异数分析。

第一个 MANOVA 指令中：H= A 规定 F 检定的分子是主效果 A。E=B(A) 规定 F 检定的分母是镶嵌效果 B(A)。删除号 (/) 后的两个选项要求 ANOVA 程序印出 H 矩阵 (导源于 F 检定的分子，在本例中即是主效果 A) 以及 E 矩阵 [导源于 F 检定的分母，在本例中是镶嵌效果 B(A)]。

第二个 MANOVA 指令中：H=B(A) 规定 F 检定的分子是镶嵌效果 B(A)。由于指令中未指明分母，故误差的平均方便自动成为此 F 检定的分母。同时，误差的矩阵也将被印出。

第三个 MANOVA 指令与第一个指令类似，但它额外地要求四对平均数的比较 (即 M= Y1-Y2, Y2-Y3, Y3-Y4, Y4-Y5)。PREFIX= DIFF 规定这四对比较的名称将分别是 DIFF1, DIFF2, DIFF3 与 DIFF4。

此外，让我们来讨论 M= 变量的转换式 这一部分的写法：

M= 转换变量 {± 转换变量}

在此，转换变量 可以是原因变量或是常数乘以原因变量。{} 中的部分可有可无。若选项 M= 中含一个以上的变量转换式，则以逗号 (,) 相隔。

选项 M= 之后也可直接以系数矩阵的横列来表示。前例中的程序若改用这种方式，则应该是：

```
MANOVA H=A E=B(A) M= (1 -1 0 0 0 ,
                        0 1 -1 0 0 ,
                        0 0 1 -1 0 ,
                        0 0 0 1 -1) PREFIX=DIFF;
```

在此必须注意的是，每一横列所含的系数必须等于因变量的数目，而且在每一横列后要加上逗号分隔。

有时，你或许希望进行趋势分析 (Trend Analysis)。下面示范此类分析的 ANOVA 程序：

```
PROC ANOVA;
CLASS GROUP;
MODEL D1-D4= GROUP;
MANOVA H= GROUP M= 3*D1-D2+D3+3*D4, D1-D2-D3-D4, -D1+3D2-3D3+D4
MNames= LINEAR QUADRTIC CUBIC / PRINTE;
```

上例 MANOVA 指令中的选项 MNames= 表明此趋势分析包含三个检定：即线性的 (一次方)、抛物线性的 (二次方)，以及 S 型的 (三次方)。其他的选项与前相同。

除了上述两个例子所示范删除号 (/) 前的 MANOVA 指令外，下面五个选项可置于

删除号 (/)之后：

(1) A PRINTH

要求印出被测效果 (即 F 检定中的分子) 的矩阵。

(2) PRINTE

要求印出 F 检定中分母的矩阵。

(3) ORTH

规定 $M =$ 所构成的平均数比较是经过标准化正交 (Orthonormalization) 的转换。

(4) CANONICAL

对 H 与 E 矩阵执行典型分析(此分析的结果会与另一统计程序 PROC CANDISC 类似), 并印出分析的结果。

(5) SUMMARY

印出每一因变量的变异数分析摘要表。 如果与选项 $M =$ 联用时, 此指令印出 M 矩阵中所转换的每一变量的变异数分析摘要表。

指令 #9 REPEATED 重复变量的名称 组数 (组名) 变量的转换 / 选项串;

假设有三种实验在四个不同的时间进行, 则每一位受试有十二个分数。假如这十二个分数分别以 Y1-Y12 表示, 则下面的指令可代表这十二个分数的统计分析:

```
REPEATED TRIAL 3 (A B C), TIME 4 (T1 T2 T3 T4);
```

这个指令言简意赅的说明了下列的数据结构:

因变量	Y1	Y2	Y3	Y4	Y5	Y6	Y7	Y8	Y9	Y10	Y11	Y12
TRIAL 的值	1	1	1	1	2	2	2	2	3	3	3	3
TIME 的值	1	2	3	4	1	2	3	4	1	2	3	4

现在让我们利用这个例子来解释 REPEATED 指令的写法:

重复变量的名称

即上例中的 TRIAL 及 TIME。若有两个以上重复变量, 则第一个重复变量的组数应是最少的, 重复变量必须与因变量有关。重复变量的名称不可以和输入文件内任何变量的名称相同, 它的长度也不可超过八个字母。

组数

界定上述重复变量的组数 (若该变量的组数为 1 时, 可以省略此选项)。从上面的例子, 我们可看出重复变量 TRIAL 有三组, 而 TIME 有四组, 所以它们的排列组合共产生十二个分数 (以 Y1-Y12 表之)。

(组名)

这个选项的值必须包含在括号内。括号内的值用来标明组别, 其个数须与组数吻合。如 TRIAL 这个重复变量有三组, 即 A, B 与 C。组名与组名之间应以空格分隔, 如: TRIAL(A B C)。

变量的转换

下面的变量转换均以 1 个自由度为原则:

(1) A POLYNOMIAL

产生多项式的正交比较。如：直线式、抛物线式，及三次方式的比较。

(2) HELMERT

比较同一变量内一组平均数与其后各组平均数的平均。如 TRIAL 中，比较 A 组平均数及 B, C 两组平均数的平均。

(3) PROFILE

比较同一变量内相邻两组的平均数。如 TRIAL 中，比较 A 与 B, B 与 C 的平均数。

(4) CONTRAST (参考组之组名或组别)

读者先选定变量中的某一组为参考组 (其组名或组别出现在此选项的括号内)，然后其他各组依序与此参考组做比较。参考组的内设值是最后一组。如：CONTRAST (A) 表示 A 是参考组，所以比较 A 与 B, A 与 C 的平均数。然而 CONTRAST (2) 则表示第二组与其他组的平均数一一作比较。

(5) MEAN (参考组之组名或组别)

比较同一变量内某一组平均数与其他各组的平均数，但不比较参考组平均数与其他各组平均数的平均。参考组的内设值是最后一组。如：MEAN (C) 表示 TRIAL 变量下的 C 组是参考组；所以比较 A 组平均数与 B, C 两组平均数的平均；以及 B 组平均数与 A, C 两组平均数的平均。但不比较 C 组平均数与 A, B 两组平均数的平均。又如：MEAN (2) 则表示第二组是参考组，因此，不比较第二组平均数与其他组平均数的平均。

请读者注意：指令中若含一个以上的重复变量，则以逗号分隔这些变量。每一变量内的资料，如：名称，组数，(组别)，变量的转换，是以空格分隔的。

删除号 (/) 后的选项，有下列八个：

(1) A NOM

不印出多变量变异数分析的结果，只印出单变量变异数分析的结果。

(2) NOU

与上述选项相反——不印出单变量变异数分析的结果，只印出多变量变异数分析的结果。

(3) PRINTM

印出变量转换的 M 矩阵 (详情请见第 31 章 GLM 程序的介绍)。

(4) PRINTH

印出多变量变异数分析的分子矩阵。

(5) PRINTRE

印出多变量变异数分析的分母矩阵。当统计假设之间彼此不独立时，此选项同时进行球形假设 (Sphericity) 的检定。

(6) PRINTV

印出每一个多变量检定的特性根与特性向量。

(7) SUMMARY

印出每一个变量转换式的变异数分析摘要表。

(8) CANONICAL

针对因变量所导出的 H 与 E 矩阵进行典型分析，其分析结果与 PROC CANDISC 的结果相似。

指令 #10 BY 变量名称串：

SAS 依据此指令所列举的变量将文件分成几个小的文件，然后对每一个小的文件分别执行 ANOVA 分析。当读者选用此指令时，文件内的数据必须先按照 BY 变量串的值做由小到大的重新排列。这个步骤可藉 PROC SORT 达成。

26.5 范 例

例一：一因子的单变量变异数分析与平均数的比较

本文件 (CLOVER) 包含一个自变量及一个因变量。自变量是苜蓿的培养基，下分六种 (即：3DOK1, 3DOK4, 3DOK5, ..., COMPOS 等)，因变量是红色苜蓿内氮气的含量。由于这是一个平衡的实验设计 (每一培养基组内有四株苜蓿)，我们用 ANOVA 程序执行单因子变异数分析，并比较各培养基组的平均氮气含量。

程 序

```
DATA CLOVER;
    INPUT STRAIN $ NITROGEN @@;
    CARDS;
3DOK1  19.4 3DOK1  32.6 3DOK1  27.0 3DOK1  32.1 3DOK1  33.0
3DOK5  17.7 3DOK5  24.8 3DOK5  27.9 3DOK5  25.2 3DOK5  24.3
3DOK4  17.0 3DOK4  19.4 3DOK4  9.1 3DOK4  11.9 3DOK4  15.8
3DOK7  20.4 3DOK7  21.0 3DOK7  20.5 3DOK7  18.8 3DOK7  18.6
3DOK13 14.3 3DOK13 14.4 3DOK13 11.8 3DOK13 11.6 3DOK13 14.2
COMPOS 17.3 COMPOS 19.4 COMPOS 19.1 COMPOS 16.9 COMPOS 20.8
;
PROC ANOVA;
    CLASS STRAIN;
    MODEL NITROGEN=STRAIN;
    MEANS STRAIN / LSD TUKEY CLDIFF;
RUN;
```

结 果

六组含氮量经 F 检定后证明不尽相同 ($P < 0.0001$)。LSD 与 TUKEY 事后检定均发现 3DOK1 组与其他五组的含氮量有显著的不同。3DOK5 组与 3DOK4, 3DOK13 两组亦有显著的不同。3DOK7 组与 3DOK5 组的平均数近似，3DOK4 与 3DOK13 两组的结果亦十分接近。其他平均数的比较在这两个事后检定的过程里有出入，不能达成共识。

报表 26.1 一因子的单变量变异数分析与平均数的比较

Analysis of Variance Procedure						
Class Level Information						
Class	Levels	Values				
STRAIN	6	3DOK1	3DOK13	3DOK4	3DOK5	3DOK7 COMPOS
Number of observations in data set = 30						
Analysis of Variance Procedure						
Dependent Variable: NITROGEN						
Source	DF	Sum of Squares	Mean Square	F Value	Pr > F	
Model	5	847.0416667	169.4083333	14.39	0.0001	
Error	24	282.5320000	11.7721667			
Corrected Total	29	1129.5736667				
R-Square		C.V.	Root MSE	NITROGEN Mean		
0.749877		17.26174	3.431059	19.8766667		
Dependent Variable: NITROGEN						
Source	DF	Anova SS	Mean Square	F Value	Pr>F	
STRAIN	5	847.0416667	169.4083333	14.39	0.0001	
Analysis of Variance Procedure						
T tests (LSD) for variable: NITROGEN						
NOTE: This test controls the type I comparisonwise error rate n the experimentwise error rate.						
Alpha= 0.05Confidence= 0.95df= 24MSE= 11.77217						
Critical Value of T= 2.06390						
Least Significant Difference= 4.4786						
Comparisons significant at the 0.05 level are indicated by '***'.						
STRAIN Comparison	Lower Confidence Limit	Difference Between Means	Upper Confidence Limit			
3DOK1 - 3DOK5	0.361	4.840	9.319	***		
3DOK1 - 3DOK7	4.481	8.960	13.439	***		
3DOK1 - COMPOS	5.641	10.120	14.599	***		
3DOK1 - 3DOK4	9.701	14.180	18.659	***		
3DOK1 - 3DOK13	11.081	15.560	20.039	***		
3DOK5 - 3DOK1	-9.319	-4.840	-0.361	***		
3DOK5 - 3DOK7	-0.359	4.120	8.599			
3DOK5 - COMPOS	0.801	5.280	9.759	***		
3DOK5 - 3DOK4	4.861	9.340	13.819	***		
3DOK5 - 3DOK13	6.241	10.720	15.199	***		
3DOK7 - 3DOK1	-13.439	-8.960	-4.481	***		
3DOK7 - 3DOK5	-8.599	-4.120	0.359			
3DOK7 - COMPOS	-3.319	1.160	5.639			
3DOK7 - 3DOK4	0.741	5.220	9.699	***		
3DOK7 - 3DOK13	2.121	6.600	11.079	***		
COMPOS - 3DOK1	-14.599	-10.120	-5.641	***		
COMPOS - 3DOK5	-9.759	-5.280	-0.801	***		
COMPOS - 3DOK7	-5.639	-1.160	3.319			
COMPOS - 3DOK4	-0.419	4.060	8.539			
COMPOS - 3DOK13	0.961	5.440	9.919	***		
3DOK4 - 3DOK1	-18.659	-14.180	-9.701	***		
3DOK4 - 3DOK5	-13.819	-9.340	-4.861	***		
3DOK4 - 3DOK7	-9.699	-5.220	-0.741	***		
3DOK4 - COMPOS	-8.539	-4.060	0.419			
3DOK4 - 3DOK13	-3.099	1.380	5.859			
3DOK13 - 3DOK1	-20.039	-15.560	-11.081	***		
3DOK13 - 3DOK5	-15.199	-6.600	-2.121	***		
3DOK13 - COMPOS	-9.919	-5.440	-0.961	***		

3DOK13 - 3DOK4 -5.859 -1.380 3.099

Tukey's Studentized Range (HSD) Test for variable: NITROGEN
NOTE: This test controls the type I experimentwise error rate.
Alpha= 0.05Confidence= 0.95df= 24MSE= 11.77217

Critical Value of Studentized Range= 4.373

Minimum Significant Difference= 6.7095

Comparisons significant at the 0.05 level are indicated by '***'.
SimultaneousSimultaneous

STRAIN Comparison		Lower Confidence Limit	Difference Between Means	Upper Confidence Limit	
3DOK1	- 3DOK5	-1.869	4.840	11.549	
3DOK1	- 3DOK7	2.251	8.960	15.669	***
3DOK1	- COMPOS	3.411	10.120	16.829	***
3DOK1	- 3DOK4	7.471	14.180	20.889	***
3DOK1	- 3DOK13	8.851	15.560	22.269	***
3DOK5	- 3DOK1	-11.549	-4.840	1.869	
3DOK5	- 3DOK7	-2.589	4.120	10.829	
3DOK5	- COMPOS	-1.429	5.280	11.989	
3DOK5	- 3DOK4	2.631	9.340	16.049	***
3DOK5	- 3DOK13	4.011	10.720	17.429	***
3DOK7	- 3DOK1	-15.669	-8.960	-2.251	***
3DOK7	- 3DOK5	-10.829	-4.120	2.589	
3DOK7	- COMPOS	-5.549	1.160	7.869	
3DOK7	- 3DOK4	-1.489	5.220	11.929	
3DOK7	- 3DOK13	-0.109	6.600	13.309	
COMPOS	- 3DOK1	-16.829	-10.120	-3.411	***
COMPOS	- 3DOK5	-11.989	-5.280	1.429	
COMPOS	- 3DOK7	-7.869	-1.160	5.549	
COMPOS	- 3DOK4	-2.649	4.060	10.769	
COMPOS	- 3DOK13	-1.269	5.440	12.149	
3DOK4	- 3DOK1	-20.889	-14.180	-7.471	***
3DOK4	- 3DOK5	-16.049	-9.340	-2.631	***
3DOK4	- 3DOK7	-11.929	-5.220	1.489	
3DOK4	- COMPOS	-10.769	-4.060	2.649	
3DOK4	- 3DOK13	-5.329	1.380	8.089	
3DOK13	- 3DOK1	-22.269	-15.560	-8.851	***
3DOK13	- 3DOK5	-17.429	-10.720	-4.011	***
3DOK13	- 3DOK7	-13.309	-6.600	0.109	
3DOK13	- COMPOS	-12.149	-5.440	1.269	
3DOK13	- 3DOK4	-8.089	-1.380	5.329	

例二：区间实验设计（又是重复观察设计）

这个实验设计中，变量 BLOCK 代表每一个观察体，也是一个区间。A 代表不重复的变量，而 B 代表重复观察的变量。

程 序

```
DATA SPLIT;
  INPUT BLOCK 1 A 2 B 3 RESPONSE;
  CARDS;
142 40.0
141 39.5
112 37.9
111 35.4
121 36.7
```

```
122 38.2
132 36.4
131 34.8
221 42.7
222 41.6
212 40.3
211 41.6
241 44.5
242 47.6
231 43.6
232 42.8
;
PROC ANOVA;
    CLASS BLOCK A B;
    MODEL RESPONSE=BLOCK A BLOCK*A B A*B;
    TEST H=A E=BLOCK*A;
    TITLE 'SPLIT PLOT DESIGN';
RUN;
```

结 果

虽然整个分析的结果达到 0.05 的显著程度,但这个显著的差异几乎全是 BLOCK 的差异造成的 ($P<0.0014$)。自变量 A, 重复观察变量 B, 或两者间的交互作用则未达 0.05 的显著程度。

报表 26.2 区间实验设计 (又是重复观察设计)

SPLIT PLOT DESIGN						
Analysis of Variance Procedure						
Class Level Information						
Class	Levels	Values				
BLOCK	2	1 2				
A	4	1 2 3 4				
B	2	1 2				
Number of observations in data set = 16						
Analysis of Variance Procedure						
Dependent Variable: RESPONSE						
Source	DF	Sum of Squares	Mean Square	F Value	Pr > F	
Model	11	182.0200000	16.5472727	7.85	0.0306	
Error	4	8.4300000	2.1075000			
Corrected Total	15	190.4500000				
R-Square		C.V.	Root MSE	RESPONSE Mean		
0.955736		3.609007	1.451723	40.2250000		
Dependent Variable: RESPONSE						
Source	DF	Anova SS	Mean Square	F Value	Pr > F	
BLOCK	1	131.1025000	131.1025000	62.21	0.0014	

A	3	40.1900000	13.3966667	6.36	0.0530
BLOCK*A	3	6.9275000	2.3091667	1.10	0.4476
B	1	2.2500000	2.2500000	1.07	0.3599
A*B	3	1.5500000	0.5166667	0.25	0.8612
Tests of Hypotheses using the Anova MS for BLOCK*A as an error term					
Source	DF	Anova SS	Mean Square	F Value	Pr > F
A	3	40.19000000	13.39666667	5.80	0.0914

例三：拉丁方格的重复实验设计

本文件 (BEETS) 是由 Smith, W. G. (1951) 发表的。文件的数据来自六种甜菜 (VARIETY)，这六种甜菜构成拉丁方格设计的行 (COL) 与列 (ROW)。另外，数据由两次收成(HARVEST) 累积而成，故形成一个拉丁方格的重复实验设计。

程 序

```

TITLE 'Latin-Square Design';
DATA BEETS;
    DO HARVEST=1 TO 2;
        DO ROW=1 TO 6;
            DO COL=1 TO 6;
                INPUT VARIETY Y @;OUTPUT;
            END;
        END;
    END;
CARDS;
3 19.1 6 18.3 5 19.6 1 18.6 2 18.2 4 18.5
6 18.1 2 19.5 4 17.6 3 18.7 1 18.7 5 19.9
1 18.1 5 20.2 6 18.5 4 20.1 3 18.6 2 19.2
2 19.1 3 18.8 1 18.7 5 20.2 4 18.6 6 18.5
4 17.5 1 18.1 2 18.7 6 18.2 5 20.4 3 18.5
5 17.7 4 17.8 3 17.4 2 17.0 6 17.6 1 17.6
3 16.2 6 17.0 5 18.1 1 16.6 2 17.7 4 16.3
6 16.0 2 15.3 4 16.0 3 17.1 1 16.5 5 17.6
1 16.5 5 18.1 6 16.7 4 16.2 3 16.7 2 17.3
2 17.5 3 16.0 1 16.4 5 18.0 4 16.6 6 16.1
4 15.7 1 16.1 2 16.7 6 16.3 5 17.8 3 16.2
5 18.3 4 16.6 3 16.4 2 17.6 6 17.1 1 16.5
;
PROC ANOVA;
    CLASS COL ROW VARIETY HARVEST;
    MODEL Y=ROW COL VARIETY ROW*COL*VARIETY
           HARVEST HARVEST*ROW
           HARVEST*VARIETY;
    TEST H=ROW COL VARIETY E=ROW*COL*VARIETY;
    TEST H=HARVEST E=HARVEST*ROW;
RUN;

```

结果

比较令人关心的甜菜种类，以及前后两季的收获量在本实验设计内分别用两个不同的分母来检定。结果显示：甜菜的种类与产量有关 ($P=0.001$)，而且收成的季节也与产量有关 ($P=0.0015$)。

报表 26.3 拉丁方格的重复实验设计

Latin-Square Design					
Analysis of Variance Procedure					
Class Level Information					
Class	Levels	Values			
COL	6	1	2	3	4 5 6
ROW	6	1	2	3	4 5 6
VARIETY	6	1	2	3	4 5 6
HARVEST	2	1	2		
Number of observations in data set = 72					
Dependent Variable: Y					
Source	DF	Sum of Squares	Mean Square	F Value	Pr > F
Model	46	98.91472222	2.15032005	7.22	0.0001
Error	25	7.44847222	0.29793889		
Corrected Total	71	106.36319444			
R-Square C.V. Root MSE Y Mean					
0.929971 3.085524 0.545838 17.6902778					
Analysis of Variance Procedure					
Dependent Variable: Y					
Source	DF	Anova SS	Mean Square	F Value	Pr > F
ROW	5	4.32069444	0.86413889	2.90	0.0337
COL	5	1.57402778	0.31480556	1.06	0.4075
VARIETY	5	20.61902778	4.12380556	13.84	0.0001
COL*ROW*VARIETY	20	3.25444444	0.16272222	0.55	0.9144
HARVEST	1	60.68347222	60.68347222	203.68	0.0001
ROW*HARVEST	5	7.71736111	1.54347222	5.18	0.0021
VARIETY*HARVEST	5	0.74569444	0.14913889	0.50	0.7729
Analysis of Variance Procedure					
Dependent Variable: Y					
Tests of Hypotheses using the Anova MS for COL*ROW*VARIETY as an error term					
Source	DF	Anova SS	Mean Square	F Value	Pr > F
ROW	5	4.32069444	0.86413889	5.31	0.0029
COL	5	1.57402778	0.31480556	1.93	0.1333
VARIETY	5	20.61902778	4.12380556	25.34	0.0001
Tests of Hypotheses using the Anova MS for ROW*HARVEST as an error term					
Source	DF	Anova SS	Mean Square	F Value	Pr > F
HARVEST	1	60.68347222	60.68347222	39.32	0.0015

第 27 章 变异数成份的分解：统计程序 PROC

VARCOMP

27.1 PROC VARCOMP 程序概述

PROC VARCOMP 旨在估计一般线性模型里变异数成分的值。这个估计的结果有助读者在随机效果 (Random Effects) 的实验设计里，执行正确的 F 检定。

■ 随机效果的实验设计

所谓随机效果的设计，系指一个自变量在母群中的组数远超过样本中读者考虑的组数。也就是说，在实际的样本中，读者只取了母群中所有可能之实验组的随机样本而已。比方说：年龄是一个自变量，若在实验中，我们取了老、中、少三个组别，则这就是一个随机效果的实验设计。因为年龄是一个连续变量，而读者强制地把这个变量分成有限的三个组别。反过来说，如果性别是一个自变量，而且实验中我们分别取了男、女两组，则这就不是一个随机的效果。因为母群中性别变量的组数 (两组) 等于样本实验中的组数 (两组)。所以，男女的差异应属于固定效果 (Fixed Effects)。

■ 基本假设

VARCOMP 程序的基本假设是其 MODEL 指令中所列举的各项效果 (如：主效果、交互效果、或镶嵌效果) 都是随机的效果。然而读者若想同时考虑固定效果与随机效果 (如：实验设计中所含的两个自变量分别是性别与年龄)，则必须在 MODEL 指令上先将这些固定效果列举出来。有关这种混合效果 (Mixed Effects) 的实验设计写法，请见 MODEL 指令中的选项 FIXED=n。

■ 四种估计变异数成份的方法

在 VARCOMP 程序中，读者可选用下列四种方法来估计变异数的成分：

(1) 一型估计法

一型估计法利用第一型离差平方和的观念 (见第 32 章的说明)，首先计算出各效果的平均方 (MS)，然后进行 F 检定。其所对应的选项是 METHOD=TYPE1。

(2) MIVQUE0 法

MIVQUE0 法由三位统计学家 Hartley, Rao, 以及 LaMotte (1978) 共同提出。这个方法所估计出来的变异数不受线性模型中固定效果的影响，是所谓的「局部最佳二次不偏估计值」(Locally Best Quadratic Unbiased Estimates)。本法与上述一型估计法类似；唯一不同的是当使用 MIVQUE0 法时，随机效果的变异数成分只针对固定效果做调整。如此，随机效果的变异数成分和固定效果的变异数成分

是彼此独立的。所以 MIVQUE0 法的计算过程非常节省时间，是 VARCOMP 程序的内置值，其所对应的选项是 METHOD=MIVQUE0。

(3) 最大可能率方法 (The Maximum Likelihood Ratio Method)

此法所导出的变异数估计值称为最大可能率估计值 (Maximum Likelihood Estimates)。这个方法的理论基础由 Hemmerle 与 Hartley (1973) 建立。估计的过程是循环式的。最原始的估计值由 MIVQUE0 法提供，然后估计值不断的被改善、修正直到其目标函数值 (即对数可能率的值) 稳定下来为止。其所对应的选项是 METHOD=ML。

(4) 局限性最大可能率法

(The Restricted Maximum Likelihood Ratio Method)

此法与上述最大可能率法十分类似，其理论基础同是最大可能率方法。两法相异之处在于局限性最大可能率法首先将线性模型中的随机效果与固定效果分开，然后进行循环式的估计直到随机效果所对应的目标函数值稳定下来为止；然而最大可能率法并不区分模型中的随机及固定效果。局限性最大可能率法所对应的选项是 METHOD=REML。

27.2 如何撰写 PROC VARCOMP 程序

PROC VARCOMP 含四道指令，它们的格式如下：

```
PROC VARCOMP 选项串;
    CLASS    自变量名称串;
    MODEL    因变量名称串=效果名称串 /选项;
    BY       变量名称串;
```

指令 #1 PROC VARCOMP 选项串;

有四个选项 (见下页)：

(1) METHOD=TYPE1 或

METHOD=MIVQUE0 或

METHOD=ML 或

METHOD=REML

这四种方法已在前面 (见本章 27.1 节) 介绍过，因此不再赘述。若省略此选项则内置值是 METHOD=MIVQUE0，因其计算非常省时。

(2) MAXITER=正整数

界定最大可能率法 (ML 法) 或局限性最大可能率法 (REML 法) 中循环的最高次数。内置值等于 50。

(3) EPSILON= 小的正实数

界定 ML 法或 REML 法里目标函数稳定度的标准。内置值是 10 的 -8 次方。如果前后两次循环所产生的函数值之差小于此数，则我们说目标函数已达稳定。

(4) DATA=输入文件名称

指明到底对那一个文件执行分析。若省略此选项，则 SAS 会自动找出在此程序之前最后形成的 SAS 文件，并对它执行分析。

指令 #2 CLASS 自变量名称串：

此指令列举所有自变量的名称。自变量可以是数值变量 (如：男=1，女=2)，也可以是文字变量 (如：男=MALE，女=FEMALE)。若是文字变量，则其变量长度值必须在十六个字母之内。若是数值变量，其值不受 16 个字母长度的限制，而且不一定要是整数。

指令 #3 MODEL 因变量名称串=效果名称串 / 选项：

因变量的数目可以是一个也可以是一个以上。当 MODEL 指令含一个以上的因变量时，VARCOMP 程序针对每一个因变量，逐一执行分析。

效果名称串可包含主效果，交互效果，以及镶嵌效果。有关这些效果的定义，读者可参阅第 26 章 ANOVA 程序的介绍。

删除号 (/) 后的选项有一个：

(1) FIXED=n (大于 0 的正整数，如：3)

这个选项规定 MODEL 指令等号右边的前几 (3) 个效果是固定效果，其后的效果则都是随机效果。

指令 #4 BY 变量名称串：

VARCOMP 程序依据此指令所列举的变量将文件分成几个小的文件，然后对每一个小的文件分别执行分析。当读者选用此指令时，文件内的数据必须先按照 BY 变量串的值作由小到大的重新排列，这个步骤可藉 PROC SORT 达成。

27.3 范 例

例一：四种估计变异数成份的方法

本文件 (A) 的数据由 Hemmerle 与 Hartley (1973) 所提供。自变量有二，即 A 与 B；因变量是 Y。三个效果中，只有 A 的主效果是固定的。B 的主效果及 A*B 的交互效果都是随机的。数据经 PROC VARCOMP 分析四次，每一次以一个不同的方法来分解变异数的成分。

程 序

```
DATA A;
    INPUT A B Y @@;
    CARDS;
1 1 237 1 1 254 1 1 246 1 2 178 1 2 179 2 1 208
2 1 178 2 1 187 2 2 146 2 2 145 2 2 141 3 1 186
3 1 183 3 2 142 3 2 125 3 2 136
;
```

```
PROC VARCOMP METHOD=TYPE1;
  CLASS A B;
  MODEL Y=A|B / FIXED=1;
PROC VARCOMP METHOD=MIVQUEO;
  CLASS A B;
  MODEL Y=A|B / FIXED=1;
PROC VARCOMP METHOD=ML;
  CLASS A B;
  MODEL Y=A|B / FIXED=1;
PROC VARCOMP METHOD=REML;
  CLASS A B;
  MODEL Y=A|B / FIXED=1;
RUN;
```

结果

第一种分析方法显示 A 的固定效果与 B 的随机效果对 Y 变异数的解释达显著的程度，A*B 的效果则不显著。

第二、三、与四法分析的过程不再检视 A 效果的显著性，只对 B 及 A*B 两个随机效果作分析。三种分析结果均显示 B 效果是显著的，然而 A*B 的交互效果则未达显著程度。

报表 27.1 四种估计变异数成份的方法

Variance Components Estimation Procedure

Class Level Information

Class	Levels	Values
A	3	1 2 3
B	2	1 2

Number of observations in data set = 16

Variance Components Estimation Procedure

Dependent Variable: Y

Source	DF	Type I SS	Type I MS	Expected Mean Square
A	2	11736.43750000	5868.21875000	$\text{Var}(\text{Error}) + 2.725 \text{ Var}(A*B) + 0.1 \text{ Var}(B) + Q(A)$
B	1	11448.12564103	11448.12564103	$\text{Var}(\text{Error}) + 2.6308 \text{ Var}(A*B) + 7.8 \text{ Var}(B)$
A*B	2	299.04102564	149.52051282	$\text{Var}(\text{Error}) + 2.5846 \text{ Var}(A*B)$
Error	10	786.33333333	78.63333333	$\text{Var}(\text{Error})$
Corrected Total	15	24269.93750000		

Variance Component	Estimate
--------------------	----------

Var(B) 1448.37683150

Var(A*B) 27.42658730

Var(Error) 78.63333333

MIVQUE(0) Variance Component Estimation Procedure

SSQ Matrix

Source	B	A*B	Error	Y
B	60.84000000	20.52000000	7.80000000	89295.38000
A*B	20.52000000	20.52000000	7.80000000	30181.30000
Error	7.80000000	7.80000000	13.00000000	12533.50000

Estimate

Variance Component Y

Var(B) 1466.12301587

Var(A*B) -35.49170274

Var(Error) 105.73659674

Maximum Likelihood Variance Components Estimation Procedure

Dependent Variable: Y

Iteration	Objective	Var(B)	Var(A*B)	Var(Error)
0	78.38503712	1031.49069751	0	74.39097179
1	78.26370438	732.36064536	0	77.40116882
2	78.26354712	723.68674709	0	77.53017748
3	78.26354712	723.66583653	0	77.53049269

Convergence criteria met.

Asymptotic Covariance Matrix of Estimates

	Var(B)	Var(A*B)	Var(Error)
Var(B)	537826.14593	0	-107.3390452
Var(A*B)	0	0	0
Var(Error)	-107.3390452	0	858.71104234

Restricted Maximum Likelihood Variance Components Estimation Procedure

Dependent Variable: Y

Iteration	Objective	Var (B)	Var (A*B)	Var (Error)
0	63.41341449	1269.52701231	0	91.55811913
1	63.04468698	1601.84198824	32.76324172	76.93555625
2	63.03115305	1468.82931677	27.22581866	78.75482763
3	63.03112651	1464.33645861	26.95640530	78.84314765
4	63.03112651	1464.36727374	26.95885252	78.84238988

Convergence criteria met.

Asymptotic Covariance Matrix of Estimates

	Var (B)	Var (A*B)	Var (Error)
Var (B)	4401703.838	1.294	-273.397
Var (A*B)	1.294	3559.113	-502.852
Var (Error)	-273.397	-502.852	1249.699

第 28 章 混合式模型的变异数分析：统计程序 PROC

MIXED

28.1 PROC MIXED 程序概述

MIXED 程序旨在执行混合式模型的线性分析，混合式模型指模型中含固定效果 (Fixed Effect) 以及随机效果 (Random Effect)，是一般线性模型的延续，因此功能更强，更广泛。

最适用于 MIXED 程序的实验设计含：镶嵌式的实验设计、重复观察的实验设计、随机效果的实验设计等。此外，MIXED 程序有如下的功能：

- (甲) 接纳各式的共变异数矩阵的结构，如双对称 (Compound Symmetry)，简易随机效果 (Simple Random Effects)，不规则形的 (Unstructured)，时间序列下的自回归型 [AR(1)]，杜氏型 (Toeplitz)，以及空间型 (Spatial)。
- (乙) 自动设定 t 检定与 F 检定的分母，尤其是针对固定效果，随机效果，以及混合效果的可估计之线性组合函数。
- (丙) 可适当地处理非平衡型的实验设计。
- (丁) 可检验受试者或分组效果是否存在。
- (戊) 利用牛顿—罗福森计算系统 (Newton-Raphson algorithm) 执行最大可能率 (ML) 或局限性最大可能率 (REML) 的估计参数的方法。
- (己) 与 ANOVA 程序或 GLM 程序 (见第 31 章) 的指令大同小异。

28.2 名词解释

■何谓混合式的线性模型？

任何数学公式的模型都旨在解释一组数据内可能存在的函数关系。一般的线性模型 (用在 ANOVA 与 GLM 程序内) 是最常见的模型，其形式如下：

$$y = X\beta + e$$

在这个表示式里，y 代表一组单变量的数据， β 是待估计的参数向量，X 是模型内已知的独立变量的值，e 代表未知的随机取样的误差。一般线性模型内 β 所对应的母群参数是一个固定的效果，因此，线性模型的最终目的就是利用 β 估计的效果值来预测因变量 y 的平均数。在估计与预测的过程中，随机误差 e 的变异数是假设为一个定值，而且 e 值间彼此独立。

若将上述的线性模型推广如下式：

$$y = X\beta + Z\gamma + e$$

则这个模型功能将更广泛。读者不难猜到：这个推广的模型就是所谓的「混合式的线性模型」(以下简称混合模型)。根据此混合模型, y , X , β , 和 e 的定义与前述的一般线性模型完全相同。新添的 Z 代表另一组的独立变量, 其对应的参数向量 v 是个随机效果的向量。除此之外, e 值间的独立关系也未必成立。

混合模型的优点在于提供 y 值的变异数以及 y 值的平均数。混合模型涵盖一般线性模型, 因为后者是前者的特例。如果我们假设 v 与 e 彼此线性独立, 其平均数都等于零, 变异数矩阵各是 G 与 R , 则混合模型可简化成下式:

$$V=ZGZ'+R$$

由于 $R=\sigma^2 \times I$ (单位矩阵), $Z=0$ (零矩阵), 因此上述简化数的混合模型就是一般的线性模型。

根据上述简化的混合模型, 读者可事先界定 G 与 R 矩阵的形式, 从而导出 y 值的变异数。MIXED 程序现有的指令可界定下面几种的 G 与 R 的矩阵结构:

- 简易型 (Simple), 适用于随机效果
- 双对称型 (Compound Symmetric), 适用于重复观察的实验设计
- 时间序列型 (Time-series), 适用于自回归型 (Autoregressive) 的数据
- 空间型 (Spatial), 或作规化型 (Kriging) 的数据
- 不规则型 (Unstructured)
- 这几种型式的排列组合

由上面的简介, 读者不难明白, MIXED 程序是第五部分所介绍的变异数分析的程序中功能最强, 用途最广的程序。下面我们举一例示范这个程序的基本功能。

28.3 MIXED 程序基本功能的示范

在这一节里, 我们将示范 MIXED 程序如何轻而易举地解决镶嵌式实验中所得的数据。首先, 让我们假设这个实验设计含 A 与 B 两个独立变量。 A 变量下的组别以随机方式分配至不同的畦区, B 变量下的组别也以随机的方式分配至 A 畦区内的次畦区下。各畦区的不同在于灌溉水份的多少, 次畦区的不同则在于一种经济作物 (如稻米) 的不同品种。

一般而言, 由于畦区的面积远远大于次畦区, 而且所牵涉的变因太多, 因此统计分析的结果较能解释次畦区间产量的不同。至于畦区间产量的不同只能自分析的数据中得到初步的了解。

在分析的过程中, A 的主效果由 F 检定 ($F=A/A*BLOCK$) 得出, B 的主效果由 $F=B/B*BLOCK$ 得出, A 与 B 的交互作用则由 $F=(A*B)/(A*B*BLOCK)$ 得出。每一个 F 检定的分母都不一样。因此, 若以 ANOVA 或第 31 章的 GLM 程序来执行这三个 F 检定, 则我们必须在程序中加上几个 TEST 指令, 否则这两个程序自动选择残差的平均方值为 F 检定的分母。这个问题在 MIXED 程序中只要藉一个 RANDOM 指令就解决了, 请看下面的程序示范 [数据由 Stroup (1989) 提供, 程序中 A , B 的定义如上, BLOCK 代表每一株稻米作物]:

```

DATA SPLIT;
    INPUT BLOCK A B Y @@;
CARDS;
1 1 1 56 1 1 2 41
1 2 1 50 1 2 2 36
1 3 1 39 1 3 2 35
2 1 1 30 2 1 2 25
2 2 1 36 2 2 2 28
2 3 1 33 2 3 2 30
3 1 1 32 3 1 2 24
3 2 1 31 3 2 2 27
3 3 1 15 3 3 2 19
4 1 1 30 4 1 2 25
4 2 1 35 4 2 2 30
4 3 1 17 4 3 2 18
;
PROC MIXED;
    CLASS A B BLOCK;
    MODEL Y=A B A*B;
    RANDOM BLOCK A*BLOCK;
RUN;

```

上述的程序列举 A, B, BLOCK 为分类变量 (见 CLASS 指令), 线性模型中的效果均为固定效果 (见 MODEL 指令中等号右式的三项 A, B, A*B), 随机效果有两项: BLOCK 与 A*BLOCK (见 RANDOM 指令)。

如此设定的模型若以简化的混合模型来理解, 则 G 矩阵是一个对角线矩阵, 对角线上的值就是 BLOCK 与 A*BLOCK 对应的变异数。R 矩阵也是一个对角线矩阵, 其值代表残差的变异数。

上述 SAS 程序执行后得到如下的结果:

The SAS System		
The MIXED Procedure		
Class Level Information		
Class	Levels	Values
A	3	1 2 3
B	2	1 2
BLOCK	4	1 2 3 4

报表的第一页是一个 "Class Level Information" 表, 读者可利用此表所列举的变量名称与其组别个数来验系数数据的输入是否正确!

接下来报表印出局限性最大可能率 (REML) 的推算过程:

The MIXED Procedure			
REML Estimation Iteration History			
Iteration	Evaluations	Objective	Criterion
0	1	106.73282502	
1	1	86.68005851	0.00000000
Convergence criteria met.			

上面的报表显示只经过一次的循环，牛顿-罗福森计算系统就达收敛指标（由 Criterion=0 表示）。每一次的循环中，牛顿-罗福森计算系统取目标函数的极小值。目标函数等于 (-2) 乘以对数可能比再加上一个常数，整个函数值打印在 "Objective" 的标题下。

循环推算的结果如下所示：

The MIXED Procedure					
Covariance Parameter Estimates (REML)					
Cov Parm	Ratio	Estimate	Std Error	Z	Pr > Z
BLOCK	6.66543027	62.39583333	56.53829541	1.10	0.2698
A*BLOCK	1.64317507	15.38194444	11.79136652	1.30	0.1921
Residual	1.00000000	9.36111111	4.41287010	2.12	0.0339

变量 BLOCK 的变异数值估计为 62.40，A*BLOCK 值是 15.38，残差是 9.36，这些值均列在标题 "Estimate" 之下。在 Estimate 行之前有一行，标示为 "Ratio"——这是各估计值除以残差的结果。因此， $62.40/9.36=6.67$ (对应 BLOCK)， $15.38/9.36=1.64$ (对应 A*BLOCK)，最后的常数 1 是显而易见的，故不赘述。

"Estimate" 之后的一行代表估计值的标准误差 (Std Error)，再后一行是标准化后的估计值，以 Z 表示。每一个 Z 值都是估计值除以标准误差之后的比率。所以 BLOCK 的 Z 值等于 $62.40/56.54=1.10$ ，其统计显著度大约是 0.2698，此打印在最后一行内，其标题为 "Pr > |Z|"，依此类推。

估计值配合着混合模型对因变量 Y 的解释能力可由接下来的报表看出：

The MIXED Procedure	
Model Fitting Information for Y	
Description	Value
Observations	24.0000
Variance Estimate	9.3611
Standard Deviation Estimate	3.0596
REML Log Likelihood	-59.8809
Akaike's Information Criterion	-62.8809
Schwartz's Bayesian Criterion	-64.2165
-2 REML Log Likelihood	119.7618
Null Model LRT Chi-Square	20.0528
Null Model LRT DF	2.0000
Null Model LRT P-Value	0.0000

Observation 等于样本的大小 (24), Variance Estimate 是前面提到的残差之变异数 (9.36), Standard Deviation Estimate 是残差的标准差 ($\sqrt{9.36}=3.06$), REML Log Likelihood (-59.88), Akaike's Information Criterion (赤池氏资讯量指标=-62.88), Schwartz's Bayesian Criterion (萧氏指标=-64.2165) 等三个数字是用来比较不等模型的优劣。根据这三个指标的定义, 愈大的值 (含负值在内) 表示其对应的模型愈好。由于本例只考虑了一个混合模型, 故无须进一步讨论这些数值的优劣。

上表中最后三个值都与可能比测试 (Likelihood Ratio Test) 有关, 在此, 可能比测试是用来比较本例中考虑的混合模型与虚无模型的相对功能。所谓的虚无模型就是一般的线性模型, 其所涵盖的效果都是固定效果。因此, 其统计显著度都以残差的平方大小为 F 检定的分母。最后让我们检视这个混合模型对 A, B 主效果以及 A*B 之间交互效果的统计显著的程度:

The SAS System					
The MIXED Procedure					
Tests of Fixed Effects					
Source	NDF	DDF	Type III F	Pr > F	
A	2	6	4.07	0.0764	
B	1	9	19.39	0.0017	
A*B	2	9	4.02	0.0566	

根据分析结果, B 的主效果特别显明。换言之, 稻米的品种对产量的多少影响最大, 其次是品种与灌溉之间的交互作用 (显著度=.0566), 最后才是灌溉量的主效果 (显著度=.0764)。若读者事先设定的第一类型错误的概率值定在 10%, 则这三个效果都可算是达到统计显著的程度了。

本节示范的混合模型若在 ANOVA 程序或 GLM 程序下执行, 则需加 TEST 的指令而且 MODEL 指令下的效果也加多, 请看下面修正过的程序:

```
PROC GLM DATA=SPLIT;
  CLASS A B BLOCK;
  MODEL Y=A|B BLOCK A*BLOCK;
  TEST H=A E=A*BLOCK;
RUN;
```

经 SAS 6.09 执行后, 得如下的结果:

The SAS System					
General Linear Models Procedure					
Dependent Variable: Y					
Source	DF	Sum of Squares	Mean Square	F Value	Pr > F
Model	14	2067.583333	147.684524	15.78	0.0001
Error	9	84.250000	9.361111		
Corrected Total	23	2151.833333			

		R-Square	C. V.	Root MSE	Y Mean
		0.960847	9.896259	3.059593	30.91667
Source	DF	Type I SS	Mean Square	F Value	Pr > F
A	2	326.583333	163.291667	17.44	0.0008
B	1	181.500000	181.500000	19.39	0.0017
A*B	2	75.250000	37.625000	4.02	0.0566
BLOCK	3	1243.500000	414.500000	44.28	0.0001
A*BLOCK	6	240.750000	40.125000	4.29	0.0256
Source	DF	Type III SS	Mean Square	F Value	Pr > F
A	2	326.583333	163.291667	17.44	0.0008
B	1	181.500000	181.500000	19.39	0.0017
A*B	2	75.250000	37.625000	4.02	0.0566
BLOCK	3	1243.500000	414.500000	44.28	0.0001
A*BLOCK	6	240.750000	40.125000	4.29	0.0256
Tests of Hypotheses using the Type III MS for A*BLOCK as an error term					
Source	DF	Type III SS	Mean Square	F Value	Pr > F
A	2	326.583333	163.291667	4.07	0.0764

28.4 如何撰写 PROC MIXED 程序

PROC MIXED 含十二道指令，其中只有 PROC MIXED 和 MODEL 是必需的，不可省略。其余十道指令则可有可无，然而当实验设计含一个以上的自变量时，读者必须用 CLASS 指令标明。以下列举 PROC MIXED 的十二道指令：

不可省略	{	PROC MIXED 选项串；
必须在 MODEL 指令前	{	CLASS 自变量名称串；
不可省略	{	MODEL 因变量 = 固定效果串 / 选项串；
必须在 MODEL 指令后	{	RANDOM 随机效果串 / 选项串； REPEATED 重复观察效果串 / 选项串； PARMS 参数的初值 / 选项串；
必须在 RANDOM 指令前	{	CONTRAST '比较式的名字'固定效果的系数 随机效果的系数 / 选项串； ESTIMATE '估计值的名字'固定效果的系数 随机效果的系数 / 选项串； LSMEANS 固定效果串 / 选项串； MAKE '表格名称'OUT = 输出资料文件名称；
出现在程序任何一处 RUN；指令前	{	BY 变量名称串； ID 变量名称串；

下面是除 PROC MIXED 以外其余十一道指令的功能简介：

PROC MIXED 指令 功能简介

CLASS	标明自变量
MODEL	界定统计模型
RANDOM	宣告某些效果是随机效果，并且界定其变异数矩阵
REPEATAED	界定混合模型中 R 矩阵的结构
PARMS	界定参数的初值或范围
CONTRAST	设定统计假设的推定空间
ESTIMATE	设定上述推定空间中的一个平面
LSMEANS	计算根据通用最小误差平方法所求得平均数
MAKE	界定一个输出资料文件，内含分析结果与公式
BY	将资料文件分成几个部分，分别对其执行统计分析
ID	观察体的识别编号

以下针对这些指令作详尽的介绍：

指令 #1 PROC MIXED 选项串：

在此指令后，有下列十六个选项：

(1) DATA=输入资料文件名称

指明对那一个 SAS 资料文件执行分析。若省略此选项，则 SAS 会自动找出在此程序前最后形成 SAS 资料文件，并对它执行分析。

(2) ORDER=FREQ 或

ORDER=DATA 或

ORDER=INTERNAL 或

ORDER=FORMATTED (内设值)

界定自变量下组别的次序，这个选项和 CONTRAST 及 ESTIMATE 指令是息息相关的。

当 ORDER=FREQ 时，观察个数最多的那一组就是第一组，次多个数的组是第二组，以下类推。

适当 ORDER=DATA 时，组别是按照输入资料文件中各组第一次出现的次序而决定的。

适当 ORDER=INTERNAL 时，组别按其数字代号 (如 1,2,3 等) 由小到大排列，或按各组名称的第一个字母顺序排列 (如 HSINCHU 在 TAIPEI 之前)。

适当 ORDER=FORMATTED 时，则组别的顺序以外部的格式 (External Format) 而决定，这也是本选项的内设值。

(3) METHOD=REML (内设值) 或

METHOD=ML 或

METHOD=MIVQUE0

界定一种统计方法可用来估计共变异数的参数。当 METHOD=REML 时，MIXED

程序选用局限性最大可能率法，这也是本选项的内设值。当 METHOD=ML 时，MIXED 程序选用最大可能率法，最后一种选择 MIVQUE0 代表玛氏最小变异数二次不偏估计法 (Minimum Variance Quadratic Unbiased Estimation)。

(4) MAXITER=正整数 (如 30)

此选项界定上述估计法的最高循环次数，内设值等于 50 次。

(5) ABSOLUTE

要求任何一种统计估计法的收敛指标是绝对的。若不界定这个选项，则收敛指标的值是相对的。下面介绍 MIXED 程序所使用的三个收敛指标：CONVF，CONVG，CONVH。

(6) CONVF 或

CONVF=正实数 (如 0.00001)

这个收敛指标取前后两次循环的目标函数值，分别以 f_k , f_{k-1} 表之，求得其差额，然后除以 f_k 的绝对值以标准化。若这个收敛指标小于内设值或读者界定的正实数 (如 0.00001)，则谓循环可收敛。上述的概念若以数学式表示，则 CONVF 的意义如下：

$$\frac{|f_k - f_{k-1}|}{|f_k|} \leq \text{正式数 (内设值为 } 10^{-8} \text{ 次方)}$$

(7) CONVG 或

CONVG=正实数

这个收敛指标先取目标函数的第一次微分 (g 函数) 计算 g 函数的第 j 个值，取其最大值。最后，再除以原目标函数值 (f_k) 以标准化。这个定义若以数学式表示，则如下式：

$$\text{CONVG} = \frac{\text{MAX}_i |g_{jk}|}{|f_k|}, \quad k = \text{循环次数}$$

若 CONVG 小于或等于 10 的 -8 次方 (内设值)，或读者界定的正实数，则谓循环可收敛。

(8) CONVH 或

CONVH=正实数

这个收敛指标利用目标函数的第一 (g_k) 与第二微分 (H_k) 函数，再除以原函数值以标准化。其数学表示式如下：

$$\text{CONVH} = \frac{g_k' H_k^{-1} g_k}{|f_k|}, \quad k = \text{循环次数}$$

若 CONVH 小于或等于 10 的 -8 次方，或是读者自定的正实数，则谓循环可收敛。CONVH 指标是 MIXED 程序的内设值，换言之，若程序中不界定 CONVF，CONVG，或 ABSOLUTE 等关键字，则 MIXED 程序自动定 CONVH 为收敛指标。

然当 H_k 是一个非满秩矩阵时, H^{-1} 就不存在, 因此, CONVH 改用下面的修正公式来计算收敛指标值:

$$\frac{g_k' g_k}{|f_k|} \leq 10 \text{ 的 } -8 \text{ 次方或读者自定的正实数}$$

(9) CL

要求 MIXED 程序计算参数的 95% 的信赖区间。读者可利用下面的选项 ALPHA 来要求其它准确度的信赖区间。

(10) ALPHA=正实数

这个值相当于第一类型错误的概率值, 因此, (1-ALPHA) 代表上述 CL 信赖区间的准确度。ALPHA 的值必须是小数, 如 0.10, 其内设值等于 0.05。

(11) NOBOUND

要求 MIXED 程序对参数的估计值不受下限的限制。由于估计的参数本身代表变异数, 因此内设的下限是 0。

(12) ASYCOV

要求打印出参数估计值的变异数矩阵。此矩阵是一个不对称的正方矩阵, 等于 $-2H^{-1}$, H 是目标函数的第二微分矩阵。 $-2H^{-1}$ 也就是费契尔资讯矩阵的反矩阵 (The Inverse Fisher Information Matrix)。

(13) MMEQ

要求报表上打印出混合模型的数学公式。

(14) MMEQSOL

要求报表上打印出上述数学模型的解, 以及系数矩阵的反矩阵。

(15) NOCLPRINT

要求抑止自变量组别的打印。

(16) NOITPRINT

要求抑止循环过程的打印。

指令 #2 CLASS 自变量名称串:

这道指令也可以写成 CLASSES 自变量名称串; 此指令标明资料文件中到底那些是统计模型中的自变量。这些变量可以是数值的或文字的, 若是文字变量, 则其长度不可超过十六个字母。此指令必须置于 MODEL 指令之前。

指令 #3 MODEL 因变量=固定效果串 / 选项串;

删除号 (/) 之前的部分:(因变量=固定效果串) 要求你首先决定到底有哪些固定的效果适用于解释数据, 然后根据 26 章 (ANOVA) 第 26.3 节的原则将它写出。

请读者注意, 这里的效果必须是固定的效果, 而且因变量只有一个。然而 ANOVA 及 GLM 程序可接受两个或两个以上的因变量。

删除号 (/) 后的选项如下:

(1) NOINT

要求 MIXED 程序将截距 (常数) 的参数排除在模型之外。否则 MODEL 指令自动把截距考虑进去。

(2) CL

要求 MIXED 程序计算每一个固定效果 95% 的信赖区间。这个信赖区间是根据 t 分配而来的。读者可利用下面的选项 ALPHA= 来要求其它准确度的信赖区间。

(3) ALPHA=正实数

这个值相当于第一类型错误的概率值, 因此, $(1-\text{ALPHA})$ 代表上述 CL 信赖区间的准确度。ALPHA 的值必须是小数, 如 0.10, 其内设值等于 0.05。

(4) SOLUTION (或 S)

要求 MIXED 程序印出固定效果的解, 亦即参数的估计值。

(5) E3

要求 MIXED 程序列出每一固定效果的第叁型可估计函数值。

(6) CHISQ

要求 MIXED 程序针对上述的函数值执行卡平方检定, 卡平方检定是除了内设的 F 检定以外另一种检验固定效果大小的统计方法。

(7) CONTAIN

这个选项只适用于极繁复的随机效果之实验设计。换言之, 前面 (第 28.2 节) 所提的 Z 矩阵必须含 200 以上的直行。在这个情况下, CONTAIN 选项利用包容法 (Containment Method) 计算 F 分母自由度的近似值, 否则, 残差的自由度自动成为 F 分母的自由度。此选项可加速 MIXED 程序的执行。

若 Z 矩阵的直行数少于 200 行, 则 MIXED 程序自动执行包容法以便得到更精确的近似值。

(8) NOCONTAIN

要求 MIXED 程序一律以残差的自由度当作 F 检定之分母的自由度。由上面 CONTAIN 选项推论, NOCONTAIN 选项只影响 Z 矩阵的直行数小于 200 的数据。

(9) SINGULAR=极小的实数

设定循环估计过程中参数估计的敏感度。内设值等于 10 的 -8 次方。

(10) ZETA=极小的正实数

此选项控制第叁型可估计函数值之可估计性检定的敏感度, 内设值是 10 的 -8 次方。若实际值小于此内设值或读者自定的下限, 则 MIXED 程序自动以 0 取代之。

(11) XPVIX

要求报表上打印出 $(X_y)'V^{-1}(X_y)$ 矩阵。

(12) XPVIXI

要求报表上打印出 $(X'V^{-1}X)^{-1}$ 矩阵与其相对应之 y 的边际 (Borders)。

(13) PREDICTED (或 P)

要求报表上打印出一系列有关观察体预测值的计算结果, 包括观察体编号 (obs), 识别代号 (由 ID 指令界定)、原数据值、预测值、预测值的变异数、预测值的标准误差、95%信赖区间的上下限、以及预测误差。表的横列则由各观察体代表。在此, 观察体必须在各独立变量上有数据才可被纳入此表中。

指令 #4 RANDOM 随机效果串 / 选项串:

RANDOM 指令用来注明实验设计中的随机效果。随机效果可以是类别的或是连续的，也可以是截距。

根据第 28.2 节的模型阐释，因变量 y 的变异数 v 等于 $ZGZ'+R$ 。在此，我们假设 G 与 R 分别代表随机变量 v 与误差变量 e 的变异矩阵， Z 矩阵代表已知随机效果的高低 (量)。此外， v 与 e 的关系是线性独立的，其平均数都等于零。

下面是删除号 (/) 后读者可界定的选项：

(1) TYPE=SIM (内设值) 或

TYPE=CS 或

TYPE=UN 或

TYPE=UN(q) 或

TYPE=AR(1) 或

TYPE=TOEP 或

TYPE=TOEP(q) 或

TYPE=SP (矩阵的结构) (坐标值)

这个选项界定上述 G 共变异数矩阵的结构。表 28.1 简单地说明各矩阵结构的形式与待估计之参数的个数：

表 28.1 G 矩阵的结构形式，代号与待估计之参数个数

TYPE=	结构形式	待估计之参数值或个数	矩阵的表示式
SIM (内设值)	简单型	=1	$\sum_{i=j} = \sigma^2_{i,j} (i=j)$
CS	双对称型	=1 或 2	$\sum_{i=j} = \sigma^2_{i,j} + \sigma^2_{j,i} (i=j)$
UN	不规则型	$n(n+1)/2$ 个	$\sum_{i=j} = \sum_{j=i}$
UN(q)	局限型	$(2n-q+1)q/2$	$\sum_{i=j} = \sum_{j=i} 1 (i-j < q)$
AR(1)	自回归型	=1 或 2	$\sum_{i=j} = \sigma^2 \rho^{ i-j }$
TOEP	杜氏型 (Toeplitz)	n 个	$\sum_{i=j} = \sum_{ i-j +1}$
TOEP(q)	局限性杜氏型 (Banded Toeplitz)	q 个	$\sum_{i=j} = \sum_{ i-j +1} 1 (i-j < q)$

上表最后一行中， $1(i=j)$ 表示当 $i=j$ 时，其值是 1；当 $i \neq j$ 时，其值是 0。同理， $1(|i-j| < q)$ 表示当 $|i-j|$ 小于 q 时，其值是 1；当 $|i-j| \geq q$ 时，其值是 0。

上表中第三行“参数之个数”里，参数值等于 1 或 2 表示当变异数 σ^2 的值源自可能率的范围，则参数的值等于 1，否则参数的值等于 2。

选项的最后一种 TYPE=SP (空间型)，其语法较复杂，因为空间型的形式在 MIXED 程序中至少有六种。兹将这六种的形式简述于表 28.2 内：

表 28.2 TYPE=SP 的矩阵结构形式

六种空间型的语法	说明	参数的值	矩阵的表示式
SP (SPH) (C)	球型	1 或 2	$\sum_{i,j} = \sigma^2 [1 - (3d_{i,j}/2\rho) - (d_{i,j}^3/2\rho^3)] 1 (d_{i,j} \leq \rho)$
SP (POW) (C)	指数函数型	1 或 2	$\sum_{i,j} = \sigma^2 \rho d_{i,j}$
SP (EXP) (C)	对数函数型	1 或 2	$\sum_{i,j} = \sigma^2 [\exp(-d_{i,j}/\rho)]$
SP (GAU) (C)	常态型	1 或 2	$\sum_{i,j} = \sigma^2 [\exp(-d_{i,j}^2/\rho^2)]$
SP (LIN) (C)	线性型	1 或 2	$\sum_{i,j} = \sigma^2 (1 - \rho d_{i,j}) 1 (\rho d_{i,j} \leq 2)$
SP (LINL) (C)	对数线性型	1 或 2	$\sum_{i,j} = \sigma^2 (1 - \rho \log(d_{i,j})) 1 (\rho \log(d_{i,j}) \leq 2)$

上表“语法”一行中的 (C) 表示 (坐标值), 读者可在括号内列举一系列数值变量名称串。然后 MIXED 程序根据观察体在这些变量上的值来计算两个点之间的欧氏距离, 亦即上表中的 d_{ij} 。

下面列举几个实例以使说明选项 TYPE= 的意义:

表 28.3 G 矩阵的实例与 TYPE=的撰写

G 矩阵形式	TYPE=	实例
简单型	SIM (内设值)	$\begin{bmatrix} \sigma^2 & 0 & 0 & 0 \\ 0 & \sigma^2 & 0 & 0 \\ 0 & 0 & \sigma^2 & 0 \\ 0 & 0 & 0 & \sigma^2 \end{bmatrix}$
双对称型	CS	$\begin{bmatrix} \sigma^2 + \sigma^2_1 & \sigma^2_1 & \sigma^2_1 & \sigma^2_1 \\ \sigma^2_1 & \sigma^2 + \sigma^2_1 & \sigma^2_1 & \sigma^2_1 \\ \sigma^2_1 & \sigma^2_1 & \sigma^2 + \sigma^2_1 & \sigma^2_1 \\ \sigma^2_1 & \sigma^2_1 & \sigma^2_1 & \sigma^2 + \sigma^2_1 \end{bmatrix}$
不规则型	UN	$\begin{bmatrix} \sigma^2_{11} & \sigma_{21} & \sigma_{31} & \sigma_{41} \\ \sigma_{21} & \sigma^2_{22} & \sigma_{32} & \sigma_{42} \\ \sigma_{31} & \sigma_{32} & \sigma^2_{33} & \sigma_{43} \\ \sigma_{41} & \sigma_{42} & \sigma_{43} & \sigma^2_{44} \end{bmatrix}$
对角线局限型	UN(1)	$\begin{bmatrix} \sigma^2_1 & 0 & 0 & 0 \\ 0 & \sigma^2_2 & 0 & 0 \\ 0 & 0 & \sigma^2_3 & 0 \\ 0 & 0 & 0 & \sigma^2_4 \end{bmatrix}$
自回归型	AR(1)	$\sigma^2 \begin{bmatrix} 1 & \rho & \rho^2 & \rho^3 \\ \rho & 1 & \rho & \rho^2 \\ \rho^2 & \rho & 1 & \rho \\ \rho^3 & \rho^2 & \rho & 1 \end{bmatrix}$
杜氏型	TOEP	$\begin{bmatrix} \sigma^2 & \sigma_1 & \sigma_2 & \sigma_3 \\ \sigma_1 & \sigma^2 & \sigma_1 & \sigma_2 \\ \sigma_2 & \sigma_1 & \sigma^2 & \sigma_1 \\ \sigma_3 & \sigma_2 & \sigma_1 & \sigma^2 \end{bmatrix}$
局限性杜氏型	TOEP(2)	$\begin{bmatrix} \sigma^2 & \sigma_1 & 0 & 0 \\ \sigma_1 & \sigma^2 & \sigma_1 & 0 \\ 0 & \sigma_1 & \sigma^2 & \sigma_1 \\ 0 & 0 & \sigma_1 & \sigma^2 \end{bmatrix}$
空间型	SP(POW)(c)	$\sigma^2 \begin{bmatrix} 1 & \rho^{d_{12}} & \rho^{d_{13}} & \rho^{d_{14}} \\ \rho^{d_{21}} & 1 & \rho^{d_{23}} & \rho^{d_{24}} \\ \rho^{d_{31}} & \rho^{d_{32}} & 1 & \rho^{d_{34}} \\ \rho^{d_{41}} & \rho^{d_{42}} & \rho^{d_{43}} & 1 \end{bmatrix}$

(2) CL

要求 MIXED 程序计算每一个随机效果 95% 的信赖区间。这个信赖区间是根据 t 分配而来的。读者可利用下面的选项 ALPHA= 来要求其他准确度的信赖区间。

(3) ALPHA=正实数

这个值相当于第一类型错误的概率值，因此， $(1-\text{ALPHA})$ 代表上述 CL 信赖区间的准确度。ALPHA 的值必须是小数，如 0.01，其内设值等于 0.05。

(4) GDATA=输入资料文件名

这个选项让读者从一个 SAS 资料文件内读取 G 矩阵的元素。GDATA=所界定的输入资料文件容许读者选用两种不同的方法来呈现 G 矩阵。第一种方法是定义一个含三直行的数据文件，三直行分别以 ROW, COL, 及 VALUE 等三个变量名称来命名，其值分别代表 G 矩阵内的元素。

第二种方法直接定义一个对角线对称的正方矩阵，若按这种方法输入 G 矩阵，则输入资料文件内必然含 n 个直行与 n 个横列。 n 直行与横列的列数必须以 COL1-COLn, 及 ROW 变量名称命名之。

(5) SUBJECT=效果名称 或

SUB=效果名称

这个选项用来指认数据中的观察体效果。若读者界定此选项，则 MIXED 程序自动将 RANDOM 指令中提及的效果镶嵌在 SUBJECT 效果内。如此，G 矩阵就会沿对角线分割成大小相等的次正方矩阵。每一个次正方矩阵都与一个观察体相对应，而且，MIXED 程序假设这些次正方矩阵之间是彼此线性独立的。

(6) G

要求报表上打印出 G 矩阵内的元素。若元素值等于 0，则以空白在报表上显示出来。

(7) GI

要求报表上打印出 G 矩阵的反矩阵元素。若元素值等于 0，则以空白在报表上显示出来。

(8) SOLUTION (或 S)

要求报表上打印出随机效果的解。

指令 #5 REPEATED 重复观察效果串 / 选项串：

这个指令的目的是为了界定第 28.2 节提到的 R 矩阵，其功用与 ANOVA 程序或 GLM 程序中的 REPEATED 指令相异。

MIXED 程序中所提的重复效果必须是一个类别变量。若数据的安排是每一位受试者有多重的资料卡，则读者必须界定重复效果变量，用来代表资料收集的先后顺序。第 28.5 节范例中的例二、三、四就是根据同一组重复观察的资料而来的。

REPEATED 指令的语法与上述 RANDOM 指令类似，删除号 (/) 后面的选项有下面几个：

(1) TYPE=SIM 或

TYPE=CS 或
 TYPE=UN 或
 TYPE=UN(q) 或
 TYPE=AR(1) 或
 TYPE=TOEP 或
 TYPE=TOEP(q) 或
 TYPE=SP (矩阵的结构) (坐标值)

这个选项定义 R 共变异数矩阵的结构。读者可参考第 4 个指令 RANDOM 下同样的 TYPE 选项, 以了解各种矩阵结构的形式与语法 (参见表 28.1 与 28.2)。上述 TYPE=CS (双对称型) 的语法也可以用 INTERCEPT 及 DIAG 取代, 试看下面的例子:

```
REPEATED INTERCEPT DIAG/SUBJECT=EFFECT;
```

在这个例子里, INTERCEPT 与 DIAG 两个关键字表示随机效果不但彼此线性独立, 而且其共变异数完全相等, 矩阵内对角线上的元素 (变异数) 也完全相等。这样的矩阵结构符合双对称型 (Compound Symmetry), 也可藉 TYPE=CS 界定。如果程序中省略 REPEATED 指令, 则 R 矩阵自动被设定成对角线矩阵, 其对角线元素是 σ^2 (共变异数)。

(2) LOCAL

要求在 R 矩阵之外加上额外的 $\sigma^2 I$ 。在此 I 是一个单位矩阵, σ^2 是待估计的变异数值。LOCAL 这个选项可用来处理时间序列模型中的取样误差或空间模型中的残差。

(3) SUBJECT=效果名称 或

SUB=效果名称

这个选项用来指认数据中的观察体效果, 是 REPEATED 指令中不可少的选项。若读者界定此选项, 则 R 矩阵就会沿对角线分割成大小相等的次正方矩阵。每一个次正方矩阵都与一个观察体相对应, 而且, MIXED 程序假设这些次正方矩阵之间是彼此线性独立的。

(4) R

要求报表上打印出 R 矩阵内的第一个次正方矩阵内的元素。若元素值等于 0, 则以空白在报表上显示出来。

(5) RI

要求报表上打印出上述 R 矩阵之第一个次正方矩阵的反矩阵。
 若矩阵的元素值等于 0, 则以空白在报表上显示出来。

指令 #6 PARMS 参数的初值 / 选项串:

这个指令旨在界定共变异数矩阵的初值或初值可能存在的范围。PARMS 指令列举这些初值的先后顺序必须与 RANDOM 或 REPEATED 指令中提及的效果串相对应。

初值的撰写方法有下面几种:

m	一个固定的值
m1, m2, ..., mn	n 个不同的初值
m to n	从 m 增至 n 的一系列初值, 相邻两初值的差额为 1
m to n by I	从 m 增至 n 的一系列初值, 相邻两初值的差额为 i
m1, m2 to m3	上述语法的混合型态

下面的例子里, 我们假设三个变异数的初值各是 60, 20 与 6。

```
PROC MIXED DATA=SPLIT;
  CLASS BLOCK A B;
  MODEL Y=A B A*B;
  RANDOM BLOCK A*BLOCK;
  PARMS (60) (20) (6)/NOITER;
RUN;
```

上面例子中所使用的选项 NOITER 是为了抑止牛顿-罗福森循环估计的过程。

删除号 (/) 后面的选项有下面几个：

(1) NOITER

要求 MIXED 程序以参数的初值为固定值, 因此, 没有必要再进行牛顿-罗福森的循环估计过程。若不界定此选项, 则参数的初值被视为估计过程的启动值, 只是用来发动牛顿-罗福森的计算程序。

(2) RATIOS

这个选项界定参数与残差变异数之间的比例, 而非参数本身的值。

指令 #7 CONTRAST '比较式的名词' 固定效果的系数 | 随机效果的系数 / 选项串;

这个指令旨在进行某些平均数的比较。由于 MIXED 程序可同时接纳固定效果以及随机效果, CONTRAST 的指令在本程序中比在 ANOVA 或 GLE 程序中的功能更宽广些。

在介绍语法的细节之前, 首先让我们介绍几个示范的例子, 这些例子是根据第 28.3 节镶嵌式的实验设计而来, 其中 A 代表畦区, B 代表次畦区, BLOCK 代表每一株稻米 A 变量下分三组, B 下分两组, BLOCK 下分四株, 因变量 Y 代表稻米的产量。

例一

```
CONTRAST 'A Broad'
  A 1 -1 0  A*B .5 .5 -.5 -.5  0  0,
  A 1 0 -1  A*B .5 .5  0  0 -.5 -.5/DF=6;
```

在这个例子里, A 组的比较有二: 第一组与第二组比, 然后是第一组与第三组比。A*B 的交互作用也有二: 第一个平均数比是在 B1 与 B2 两组内, 先求得 A1 与 A2 平均数的差; 然后, 这两个差再作比较。第二个平均数比与第一个相仿, 只须将 A2 组的平均数以 A3 代替即可。

选项 DF=6 的目的是为了界定 F 检定之分母的自由度。若不界定此选项, 则

MIXED 程序自动以残差的自由度来取代。残差的自由度是内设值，在此例中，其值等于 9。单引号之内是这些比较式的名字，有关比较式的命名原则以及 Broad 这个字的意义，我们在例二之后有详细说明。

CONTRAST 'A Narrow'									
A	1	-1	0						
A*B	.5	.5	-.5	-.5	0	0	1		
A*BLOCK	.25	.25	.25	.25					
	-.25	-.25	-.25	-.25					
	0	0	0	0					
A	1	0	-1						
A*B	.5	.5	0	0	-.5	-.5	1		
A*BLOCK	.25	.25	.25	.25					
	0	0	0	0					
	-.25	-.25	-.25	-.25					

例一中的 Broad 空间表示 A 是一个随机效果，因此 CONTRAST 所得的结果可推广到 A 效果下所有可能的组别 (或畦区)。例二中的 Narrow 空间表示 A 是一个固定效果，因此 CONTRAST 的分析结果只适合用来解释现有 A 效果下的组别 (或畦区) 而非所有可能的组别。有关这两种推定空间的定义与应用，请参见 McLean, Sanders, Stroup 于 1991 年发表的论文。

从以上两个例子里，我们可以归纳出点 CONTRAST 的命名原则：

- “比较式的名字”必须放在单引号内，名字的长度以二十个字母为限。命名的方式不拘，但不可夹带分号 (;)。
- 固定效果的系数必须写在固定效果名称的后面。固定效果必须在 MODEL 指令中出现过或是关键字 INTERCEPT (表截距)。针对每一效果而言，其系数总和必须等于 0，而且只能是整数或小数。各系数之间以空格隔开。
- 随机的效果必须放在直线 (|) 之后而且必须是 RANDOM 指令中提及的效果。
- 若同一个 CONTRAST 指令内含一个以上的比较式，则以逗号 (,) 将系数串隔开。

删除号 (/) 后的选项有下面几个：

(1) DF=正整数

此选项界定 F 检定之分母的自由度。内设值是残差的自由度，也就是 N(样本数) - rank(XZ) (矩阵 XZ 的秩)。

(2) CHISQ

要求 MIXED 程序额外地执行卡平方检定。如此，报表上同时打印 F 检定与卡平方检定的结果。

(3) E

要求将 L 矩阵的元素打印在报表上。

(4) SINGULAR=极小的正实数 (如 0.00001)

界定转换函数可估计的程度，内设值等于 10 的 -4 次方。

指令 #8 ESTIMATE '估计值的名字' 固定效果的系数 | 随机效果的系数/选项串;

这个指令与上述的 CONTRAST 指令类似，它们遵循同样的原则。但除此之外，ESTIMATE 指令还打印出对 L'P 的估计值，其标准误差以及 t 检定的结果和统计显著度。每一个 ESTIMATE 指令只接受一横列的 L 向量。

请看下面示范的三个例子 (根据第 28.3 节镶嵌式的实验设计而来)：

例一

```
ESTIMATE 'A1 MEAN NARROW' INTERCEPT 1
      A 1      B .5 .5  A*B .5 .5 |
BLOCK .25 .25 .25 .25
A*BLOCK .25      0 0 .25 0 0
      .25      0 0 .25 0 0;
```

例二

```
ESTIMATE 'A1 MEAN INTERMED' INTERCEPT 1
      A 1      B .5 .5  A*B .5 .5 |
BLOCK .25 .25 .25 .25;
```

例三

```
ESTIMATE 'A1 MEAN BROAD' INTERCEPT 1
      A 1      B .5 .5
A*B .5 .5 ;
```

有关这三个例子所导出的三个推定空间的说明与报表解释，请参阅第 28.5 节的例一。删除号 (/) 后的选项列举如下：

(1) DF=正整数

此选项界定 F 检定之分母的自由度。内设值是残差的自由度，也就是 N (样本数) -rank (XZ) (矩阵 XZ 的秩)。

(2) CL

要求 MIXED 程序计算每一个估计效果 95% 的信赖区间。这个信赖区间根据 t 分配而来的。读者可利用下面的选项 ALPHA= 来决定其它准确度的信赖区间。

(3) ALPHA=正实数

这个值相当于第一类型错误的概率值，因此，(1-ALPHA) 代表上述 CL 信赖区间的准确度。ALPHA 的值必须是小数，如 0.01，其内设值等于 0.05。

(4) DIVISOR=整数

MIXED 程序用此整数当作分母来删除号前的效果系数。

(5) E

打印出线性函数的向量，L。

(6) SINGULAR=

以此数为标准，检定 ESTIMATE 指令中所导出的线性函数是否为可估计的 (Estimable)。其检定标准与前述 CONTRAST 指令中同一个选项完全一致，故不再赘述。内设值也等于 10 的 -4 次方。

指令 #9 LSMEANS 固定效果串 / 选项串:

LSMEANS 是以通用最小误差平方法所估计之平均数的代称 (英文称为 Generalized Least Squares Means)。这个估计法只用做计算固定效果的边际平均数 (Marginal Mean)，而且实验设计假设为平衡的设计。

下面示范 LSMEANS 的语法：

```
PROC MIXED;
      CLASS      A   B;
      MODEL      A   B   A*B;
      LSMEANS    A   B   ;
```

请读者注意：LSMEANS 指令里所提的效果，必须是 MODEL 指令里已经提过的固定效果。效果值之估计藉助线性函数之向量，L。L 的结构必须是可估计的，否则报表的结果以遗漏值表示。

删除号 (/) 后的选项有下面数个：

(1) DF=正整数

此选项界定 t 检定之自由度。t 检定的虚无假设是母群中对应的固定效果之平均数等于 0。

(2) CL

要求 MIXED 程序计算每一个 LSMEANS 95% 的信赖区间。这个信赖区间根据 t 分配而来。读者可利用下面的选项 ALPHA= 来要求其他准确度的信赖区间。

(3) ALPHA=正实数

这个值相当于第一类型错误的概率值，因此，(1-ALPHA) 代表上述 CL 信赖区间的准确度。ALPHA 的值必须是小数，如 0.01，其内设值等于 0.05。

(4) E

打印出线性函数的向量，L。

(5) SINGULAR=极小的正实数

以此为标准，检定 LSMEANS 指令中所导出的线性函数是否为可估计的 (Estimable)。其检定标准与前述 CONTRAST 指令中同一个选项完全一致，故不再赘述。内设值也等于 10 的 -4 次方。

指令 #10 MAKE '表格名称' OUT=输出资料文件名称:

这个指令旨在将 MIXED 程序产生的各种表格转变成 SAS 的资料文件，其文件名以 OUT= 指令界定。'表格名称' 内的名称由表 28.4 而来。请读者注意：表格的名称可用英文大小写键入，然而前后必须以单引号括住。

表 28.4 各式表格名称代号与其意义

表格名称	意义
PREDICTED	因变量 y 的预测值
ASYCOV	参数之不对称的变异数矩阵
COVPARMS	共变量参数的估计值
G	G 矩阵
GI	G 矩阵之反矩阵
CLASSLEVELS	指令 CLASS 的组别名称与个数
PARMS	指令 PARMS 的结果
CONTRAST	指令 CONTRAST 的结果
ESTIMATE	指令 ESTIMATE 的结果
LSMEANS	指令 LSMEANS 的结果
TESTS	效果检定的结果
SOLUTIONF	固定效果解的向量
SOLUTIONR	随机效果解的向量
ML	最大可能率循环估计之过程
REML	局限性最大可能率循环估计之过程
FITTING	混合型模型的适合度
MMEQ	混合型模型的数学公式
MMEQSOL	混合型模型的数学公式之解
COEFFICIENTS#	线性函数 L 的系数。若 # 以 1 取代，则所得的表格是根据 MODEL 指令中的 E3 选项。 若 # 以 2 或 3 或 4 取代，则是根据 CONTRAST，或 ESTIMATE 或 LSMEANS 指令中的 E 选项。
R	R 的矩阵的第一正方次矩阵
RI	上述次矩阵的反矩阵

指令 #11 BY 变量名称串：

SAS 依据此指令所列举的变量将资料文件分成几个小的资料文件，然后对每一个小的资料文件分别执行 MIXED 程序的分析，当读者选用此指令时，资料文件内的数据必须先按照 BY 变量串的值重新做由小到大的排列，这个步骤可藉 PROC SORT 达成。

指令 #12 ID 变量名称串：

这个指令旨在报表上识别各观察体及其估计值与估计误差。

28.5 范 例

例一：镶嵌式实验设计的再思

本例延续第 28.3 节所提到的镶嵌式实验设计的数据分析，如前所示，这个实验设计

含 A 与 B 两个独立变量。A 变量下的组别以随机方式分配至不同的畦区，B 变量下的组别也以随机方式分配至 A 畦区内的次畦区下。各畦区的不同在于灌溉水份的多少，次畦区的不同则在于经济作物——如稻米——的不同品种。下面程序中所提到的 BLOCK 变量代表每一株稻米作物。

程 序

```

TITLE 'Example 28.1: Split-plot ANALYSIS';
OPTIONS LS=75 NODATE;
DATA SPLIT;
    INPUT BLOCK A B Y @@;
CARDS;
1 1 1 56    1 1 2 41
1 2 1 50    1 2 2 36
1 3 1 39    1 3 2 35
2 1 1 30    2 1 2 25
2 2 1 36    2 2 2 28
2 3 1 33    2 3 2 30
3 1 1 32    3 1 2 24
3 2 1 31    3 2 2 27
3 3 1 15    3 3 2 19
4 1 1 30    4 1 2 25
4 2 1 35    4 2 2 30
4 3 1 17    4 3 2 18
;
PROC MIXED;
    CLASS A B BLOCK;
    MODEL Y=A|B;
    RANDOM BLOCK A* BLOCK;
RUN;

```

这个程序另一种撰写方式如下：

```

PROC MIXED;
    CLASS A B BLOCK;
    MODEL Y=A|B;
    RANDOM INTERCEPT A/SUBJECT=BLOCK;
RUN;

```

上述的撰写方式将 BLOCK (稻米) 镶嵌在 A (畦区) 内，因此可将其自 RANDOM 指令的效果中净化出来。第二种撰写程序的方式不适用于 REPEATED 指令。两种撰写方式所得的结果完全一致 (见 28.3 节的报表)。

最后，这个程序也可加插三个 ESTIMATE 指令以便执行各式的推定空间：

```

ESTIMATE 'A1 MEAN NARROW' INTERCEPT 1 A 1 B .5 .5 A*B .5 .5|
                                BLOCK .25 .25 .25 .25
                                A*BLOCK .25 .25 .25 .25 0 0 0 0 0 0 0 0;
ESTIMATE 'A1 MEAN INTERMED' INTERCEPT 1 A 1 B .5 .5 A*B .5 .5|
                                BLOCK .25 .25 .25 .25;
ESTIMATE 'A1 MEAN BROAD' INTERCEPT 1 A 1 B .5 .5 A*B .5 .5;
RUN;

```

根据这三个 ESTIMATE 指令，我们得到如下的结果：

结 果

报表 28.1 镶嵌式实验设计的再思

The MIXED Procedure					
ESTIMATE Statement Results					
Parameter	Estimate	Std Error	DDF	T	Pr > T
A1 MEAN NARROW	32.87500000	1.08172958	9	30.39	0.0000
A1 MEAN INTERMED	32.87500000	2.23955911	9	14.68	0.0000
A1 MEAN BROAD	32.87500000	4.54032855	9	7.24	0.0000

值得注意的是：三个推定空间的值完全一样 ($=32.875$)，然而标准误差从 1.08172958 增至 4.54032855。这是因为第一个推定空间是最狭窄的 (Narrow Space)，所有 t 检定的结果 (30.39) 只能反推回变量 BLOCK 以及 A*BLOCK 下实际被观察的稻米。因此，这个结果的实用价值不如其他两个结果。

第二个推定空间介于最狭窄与最宽广之间 (Intermediate Space)，其 t 检定结果 (14.68) 可反推至变量 BLOCK 下实际被观察的稻米以及 A*BLOCK 交互作用下母群内所有可能的排列组合。

第三个推定空间最宽广的 (Broad Space)，也是一般研究者最终极的母群的空间。因此，t 的结果 (7.24) 可反推是回至所有可能被观察的稻米作物以及 A*BLOCK 母群内所有可能的排列组合。由于 7.24 达到极显著 ($P \leq .00001$) 的统计度，我们可说 A 的效果的确存在。换言之，灌溉水份的多少确实会影响所有稻米的产量。

例二：重复观察的分析 (上)

下列数据来自 Pothoff 与 Roy (1964) 的论文，内含 11 位女孩与 16 位男孩在八岁、十岁、十二岁、与十四岁时身体发育的资料。这组资料在 1986 年经由 Gennrich 与 Schluchter 分为几个步骤来分析，每一次尝试不同的共变量结构的假设。

首先，我们示范不规则型的结构假设：

程 序

```

TITLE 'Example 28.2: Repeated Measures ANALYSIS';
OPTIONS LS=75 NODATE;
DATA REPEAT;
INPUT PERSON SEX $ Y1-Y4;

```

```
Y=Y1; AGE=8; OUTPUT;
Y=Y2; AGE=10; OUTPUT;
Y=Y3; AGE=12; OUTPUT;
Y=Y4; AGE=14; OUTPUT;

CARDS;
1 F 21.0 20.0 21.5 23.0
2 F 21.0 21.5 24.0 25.5
3 F 20.5 24.0 24.5 26.0
4 F 23.5 24.5 25.0 26.5
5 F 21.5 23.0 22.5 23.5
6 F 20.0 21.0 21.0 22.5
7 F 21.5 22.5 23.0 25.0
8 F 23.0 23.0 23.5 24.0
9 F 20.0 21.0 22.0 21.5
10 F 16.5 19.0 19.0 19.5
11 F 24.5 25.0 28.0 28.0
12 M 26.0 25.0 29.0 31.0
13 M 21.5 22.5 23.0 26.5
14 M 23.0 22.5 24.0 27.5
15 M 25.5 27.5 26.5 27.0
16 M 20.0 23.5 22.5 26.0
17 M 24.5 25.5 27.0 28.5
18 M 22.0 22.0 24.5 26.5
19 M 24.0 21.5 24.5 25.5
20 M 23.0 20.5 31.0 26.0
21 M 27.5 28.0 31.0 31.5
22 M 23.0 23.0 23.5 25.0
23 M 21.5 23.5 24.0 28.0
24 M 17.0 24.5 26.0 29.5
25 M 22.5 25.5 25.5 26.0
26 M 23.0 24.5 26.0 30.0
27 M 22.0 21.5 23.5 25.0
;

PROC MIXED METHOD=ML SCORING;
  CLASS PERSON SEX;
  MODEL Y=SEX AGE*SEX/NOINT S;
  REPEATED/TYPE=UN SUBJECT=PERSON R;

RUN;
```

由于 Gennrich 与 Schluchter 1986 年的分析采用最大可能率法，因此程序中界定 METHOD=ML。模式中的固定效果含截距与 AGE (年龄) 的斜率。关键字 NOINT 表示不同的截距以 SEX (性别) 的组别代表，而不是以相对于总截距的正、负距离来表示。选项 S 要求所有固定效果的解法打印在报表上。

选项 TYPE=UN 表示每一位 SUBJECT=PERSON 的变异数矩阵没有任何可寻的形式。选项 R 要求将第一位受试者 (PERSON=1, 女孩) 的矩阵估计值打印在报表上。

结 果

报表 28.2 重复观察的分析 (上)

```

Example 28.2: Repeated Measures ANALYSIS

The MIXED Procedure

Class Level Information

Class      Levels  Values
PERSON      27    1 2 3 4 5 6 7 8 9 10 11 12 13
              14 15 16 17 18 19 20 21 22 23
              24 25 26 27
SEX          2    F M

The MIXED Procedure

ML Estimation Iteration History

Iteration  Evaluations      Objective      Criterion
          -----
          0              1    279.75103669
          1              2    220.98663078    0.00000248
          2              1    220.98632822    0.00000003
          3              1    220.98632494    0.00000000

Scoring stopped after iteration 1.

Convergence criteria met.

The MIXED Procedure

```

R Matrix for PERSON 1 (第一位受试者的共变异数矩阵)

Row	COL1	COL2	COL3	COL4
1	5.11919889	2.44090159	3.61051034	2.52224364
2	2.44090159	3.92794750	2.71751364	3.06234944
3	3.61051034	2.71751364	5.97979819	3.82346068
4	2.52224364	3.06234944	3.82346068	4.61798404

Covariance Parameter Estimates (MLE)

Cov	Parm	Estimate	Std Error	Z	Pr > Z	
DIAG	UN(1,1)	5.11919889	1.41686560	3.61	0.0003	上述矩阵元素值的标准误差与
	UN(2,1)	2.44090159	0.98353460	2.48	0.0131	估计值的华氏 z 鉴定 (Waldz)。
	UN(2,2)	3.92794750	1.08245774	3.63	0.0003	所有鉴定都达 5% 的统计显著度。
	UN(3,1)	3.61051034	1.27665783	2.83	0.0047	
	UN(3,2)	2.71751364	1.07398146	2.53	0.0114	
	UN(3,3)	5.97979819	1.62787950	3.67	0.0002	
	UN(4,1)	2.52224364	1.06485905	2.37	0.0179	
	UN(4,2)	3.06234944	1.01346595	3.02	0.0025	
	UN(4,3)	3.82346068	1.25076893	3.06	0.0022	
	UN(4,4)	4.61798404	1.25732801	3.67	0.0002	
Residual		1.00000013	.	.	.	

Model Fitting Information for Y

Description	Value
Observations	108.0000
Variance Estimate	1.0000
Standard Deviation Estimate	1.0000
Log Likelihood	-209.739
Akaike's Information Criterion	-219.739
Schwartz's Bayesian Criterion	-233.149
-2 Log Likelihood	419.4770
Null Model LRT Chi-Square	58.7647 不规则的结构假设达统计显著度，
Null Model LRT DF	9.0000 较单位矩阵的结构更优越。
Null Model LRT P-Value	0.0000

Solution for Fixed Effects

Parameter	Estimate	Std Error	DDF	T	Pr > T	
SEX F	17.42536849	1.12838047	104	15.44	0.0000	女孩的截距较男孩为高
SEX M	15.84228933	0.93560366	104	16.93	0.0000	女孩发育的斜度是男孩的一半
AGE*SEX F	0.47636470	0.09541519	104	4.99	0.0000	
AGE*SEX M	0.82680330	0.07911409	104	10.45	0.0000	

The MIXED Procedure

Tests of Fixed Effects

Source	NDF	DDF	Type III F	Pr > F	
SEX	2	104	262.60	0.0000	两个固定效果均达统计显著程度
AGE*SEX	2	104	67.07	0.0000	

若读者认为身体发育的资料是年龄的函数而欲假定共变异数矩阵的结构遵循时间序列的形式，则本例程序中 REPEATED 指令可以改成下面的指令：

```
REPEATED INTERCEPT AGE/TYPE=AR(1) SUB=PERSON G;
```

例三：重复观察的分析（中）

本例沿袭上面例二的数据与问题，然而在此例中我们界定共变异数矩阵的结构是双对称型的。如此导出的结果可与例二的结果比较。

程 序

```
TITLE 'Example 28.3: Repeated Measures with Compound Symmetry';
OPTIONS LS=75 NODATE;
DATA REPEAT;
    INPUT PERSON SEX $ Y1-Y4;
    Y=Y1; AGE=8; OUTPUT;
    Y=Y2; AGE=10; OUTPUT;
    Y=Y3; AGE=12; OUTPUT;
    Y=Y4; AGE=14; OUTPUT;
CARDS;
(数据来自本章例二)
;
PROC MIXED METHOD=ML SCORING;
    CLASS PERSON SEX;
```

```
MODEL Y=SEX AGE*SEX/NOINT S;  
REPEATED/TYPE=CS SUBJECT=PERSON R;  
RUN;
```

结 果

报表 28.3 重复观察的分析 (中)

Example 28.3: Repeated Measures with Compound Symmetry

The MIXED Procedure

Class Level Information

Class	Levels	Values
PERSON	27	1 2 3 4 5 6 7 8 9 10 11 12 13 14 15 16 17 18 19 20 21 22 23 24 25 26 27
	2	F M

ML Estimation Iteration History

Iteration	Evaluations	Objective	Criterion
0	1	279.75103669	
1	1	230.14833485	0.00000000

由于数据来自平衡式试验，
故循环估计的过程一次即可收敛。

Scoring stopped after iteration 1.

Convergence criteria met.

R Matrix for PERSON 1

Row	COL1	COL2	COL3	COL4
1	4.90515835	3.03056169	3.03056169	3.03056169
2	3.03056169	4.90515835	3.03056169	3.03056169
3	3.03056169	3.03056169	4.90515835	3.03056169
4	3.03056169	3.03056169	3.03056169	4.90515835

双对称的共变异数矩阵。

Covariance Parameter Estimates (MLE)

Cov Parm	Ratio	Estimate	Std Error	Z	Pr > Z
DIAG CS	1.61664733	3.03056169	0.95520746	3.17	0.0015
Residual	1.00000000	1.87459666	0.29456445	6.36	0.0000

Model Fitting Information for Y

Description	Value	
Observations	108.0000	
Variance Estimate	1.8746	
Standard Deviation Estimate	1.3692	
Log Likelihood	-214.320	
Akaike's Information Criterion	-216.320	
Schwartz's Bayesian Criterion	-219.002	
-2 Log Likelihood	428.6391	
Null Model LRT Chi-Square	49.6027	双对称的结构假设达统计显著度,
Null Model LRT DF	1.0000	较单位矩阵的结构更优越。
Null Model LRT P-Value	0.0000	

Solution for Fixed Effects

Parameter	Estimate	Std Error	DDF	T	Pr > T	
SEX F	17.37272727	1.16152410	104	14.96	0.0000	与例二的结果类似。
SEX M	16.34062500	0.96308491	104	16.97	0.0000	
AGE*SEX F	0.47954545	0.09230869	104	5.20	0.0000	
AGE*SEX M	0.78437500	0.07653832	104	10.25	0.0000	

Tests of Fixed Effects

Source	NDF	DDF	Type III F	Pr > F	
SEX	2	104	255.79	0.0000	与例二的结构类似。
AGE*SEX	2	104	66.01	0.0000	

本例中的 SAS 程序也可改写成：

```
PROC MIXED DATA=REPEAT METHOD=ML SCORING;
  CLASS PERSON SEX;
  MODEL Y=SEX AGE*SEX/NOINT S;
  RANDOM PERSON;
RUN;
```

或是

```
PROC MIXED DATA=REPEAT METHOD=ML SCORING;
  CLASS PERSON SEX;
  MODEL Y=SEX AGE*SEX/NOINT S;
  REPEATED INTERCEPT DIAG/SUBJECT=PERSON r;
```

所得的报表结果都是正确的。

例四：重复观察的分析（下）

沿袭例二与例三的数据与分析的步骤，我们在本例中额外地界定男、女孩两组的共变异数矩阵可以是不同的，然而其结构均为双对称型。请看下面 SAS 程序的语法：

程 序

```
TITLE 'Example 28.4: Repeated Measures with HETEROGENEITY';
OPTIONS LS=75 NODATE;
DATA PEPEAT;
  INPUT PERSON SEX $ Y1-Y4;
  Y=Y1; AGE=8; OUTPUT;
  Y=Y2; AGE=10; OUTPUT;
  Y=Y3; AGE=12; OUTPUT;
  Y=Y4; AGE=14; OUTPUT;
CARDS;
1 F 21.0 20.0 21.5 23.0
2 F 21.0 21.5 24.0 25.5
3 F 20.5 24.0 24.5 26.0
4 F 23.5 24.5 25.0 26.5
5 F 21.5 23.0 22.5 23.5
6 F 20.0 21.0 21.0 22.5
7 F 21.5 22.5 23.0 25.0
8 F 23.0 23.0 23.5 24.0
9 F 20.0 21.0 22.0 21.5
10 F 16.5 19.0 19.0 19.5
11 F 24.5 25.0 28.0 28.0
```

```

12 M 26.0 25.0 29.0 31.0
13 M 21.5 22.5 23.0 26.5
14 M 23.0 22.5 24.0 27.5
15 M 25.5 27.5 26.5 27.0
16 M 20.0 23.5 22.5 26.0
17 M 24.5 25.5 27.0 28.5
18 M 22.0 22.0 24.5 26.5
19 M 24.0 21.5 24.5 25.5
20 M 23.0 20.5 31.0 26.0
21 M 27.5 28.0 31.0 31.5
22 M 23.0 23.0 23.5 25.0
23 M 21.5 23.5 24.0 28.0
24 M 17.0 24.5 26.0 29.5
25 M 22.5 25.5 25.5 26.0
26 M 23.0 24.5 26.0 30.0
27 M 22.0 21.5 23.5 25.0
;
PROC MIXED METHOD=ML SCORING;
  CLASS PERSON SEX;
  MODEL Y=SEX AGE*SEX/NOINT S;
  REPEATED INTERCEPT DIAG/ SUBJECT=PERSON GROUP=SEX;
RUN;

```

结 果

报表 28.4 重复观察的分析（下）

Example 28.4: Repeated Measures with HETEROGENEITY

Class Level Information

Class	Levels	Values
PERSON	27	1 2 3 4 5 6 7 8 9 10 11 12 13 14 15 16 17 18 19 20 21 22 23 24 25 26 27
SEX	2	F M

ML Estimation Iteration History

Iteration	Evaluations	Objective	Criterion
0	1	279.75103669	
1	1	210.32224911	0.00000000

一次循环估计就达收敛指标。

Scoring stopped after iteration 1.

Convergence criteria met.

Covariance Parameter Estimates (MLE)

Cov Parm		Estimate	Std Error	Z	Pr > Z	
INTERCEPT	SEX F	3.88038912	1.71788299	2.26	0.0239	男、女的共变异矩阵的 非对角线元素不等。
INTERCEPT	SEX M	2.44630534	1.11754866	2.19	0.0286	
DIAG	SEX F	0.59001377	0.14525135	4.06	0.0000	男、女的共变异矩阵的 对角元素不等。
DIAG	SEX M	2.75774740	0.56292283	4.90	0.0000	
Residual		1.00000000	.	.	.	

Model Fitting Information for Y

Description	Value	
Observations	108.0000	
Variance Estimate	1.0000	
Standard Deviation Estimate	1.0000	
Log Likelihood	-204.406	
Akaike's Information Criterion	-208.406	比例三的 (-216.3) 值更高, 所以此模型较好
Schwartz's Bayesian Criterion	-213.771	比例三的 (-216.-0) 值更高, 所以此模型较好
-2 Log Likelihood	408.8130	
Null Model LRT Chi-Square	69.4288	
Null Model LRT DF	3.0000	
Null Model LRT P-Value	0.0000	

Solution for Fixed Effects

Parameter	Estimate	Std Error	DDF	T	Pr > T	
SEX F	17.37272727	0.83107137	104	20.90	0.0000	估计固定效果值与 例三一样, 单标准误差 略有不同
SEX M	16.34062500	1.11299466	104	14.68	0.0000	
AGE*SEX F	0.47954545	0.05178688	104	9.26	0.0000	
AGE*SEX M	0.78437500	0.09283297	104	8.45	0.0000	

Tests of Fixed Effects

Source	NDF	DDF	Type III F	Pr > F	
SEX	2	104	326.26	0.0000	与例三的结果完全相同
AGE*SEX	2	104	78.57	0.0000	

由于赤池资讯量指标与贝氏资讯指标显示例四的模型比例三的模型更好, 我们可进一步探讨: 是否这个优劣差别达到统计显著的程度? 欲追究这个问题, 我们首先计算两个模型所导出的 -2 Log Likelihood 值: 428.64 (例三) 和 408.81 (例四) 的差 (19.83), 然后与卡平方的分配 (自由度=2) 作比较。我们不难发现, 19.83 的值已达 0.0001 的显著度。如此, 我们可下结论说: 例四的模型的确明显地好于例三

例五：可能率的推测与绘图

本例的资料文件 (MLPLOT) 与第 27 章 (VARCOMP) 的例一的 A 文件完全相同，由 Hemmele 与 Hartley (1973) 所提供。自变量有二，即 A 与 B；因变量是 Y，与烤箱的实验有关。

三个效果中，只有 A 的主效果是固定的，B 的主效果以及 A*B 的交互效果都是随机的。

程 序

```
TITLE 'Example 28.5: LLIKELIHOOD PLOTTED';
OPTIONS LS=75 NODATE;
DATA MLPLOT;
    INPUT A B Y @@;
CARDS;
1 1 237 1 1 254 1 1 246
1 2 178 1 2 179
2 1 208 2 1 178 2 1 187
2 2 146 2 2 145 2 2 141
3 1 186 3 1 183
3 2 142 3 2 125 3 2 136
;
PROC MIXED ASYCOV MMEQ MMEQSOL;
    CLASS A B;
    MODEL Y=A/PREDICTED;
    RANDOM B A*B;
    LSMEANS A;
    PARMS (17 TO 20 BY .1) (.3 TO .4 BY .005) (1.0);
    MAKE 'PARMS' OUT=PARMS;
RUN;
PROC G3D DATA=PARMS;
    PLOT COL1*COL2=REML_LL;
RUN;
```

上述程序中 ASYCOV 关键字要求最终的共变异数矩阵不必是对称的矩阵。MMEQ 与 MMEQSOL 要求报表上印出固定效果模型的形式与其解。

在 MODEL 指令下的选项 PREDICTED 表示每一观察体在因变量 Y 上之预测值会印在报表上，同时被收集在 "PARMS" 的资料文件内。

结 果

报表 28.5 可能率的推测与绘图

```
Example 28.5: LLIKELIHOOD PLOTTED
The MIXED Procedure
Class Level Information
Class      Levels  Values
A          3      1 2 3
B          2      1 2
```

Parameter Search

COL1	COL2	COL3	Variance	REML_LL	-2REML_LL	Objective
17.0000	0.3000	1.0000	80.1400	-52.4699	104.9399	81.0475
17.0000	0.3050	1.0000	80.0466	-52.4697	104.9393	81.0469
17.0000	0.3100	1.0000	79.9545	-52.4694	104.9388	81.0464
17.0000	0.3150	1.0000	79.8637	-52.4692	104.9384	81.0460
17.0000	0.3200	1.0000	79.7742	-52.4691	104.9381	81.0457
17.0000	0.3250	1.0000	79.6859	-52.4690	104.9379	81.0455
17.0000	0.3300	1.0000	79.5988	-52.4689	104.9378	81.0454
17.0000	0.3350	1.0000	79.5129	-52.4689	104.9377	81.0453
17.0000	0.3400	1.0000	79.4282	-52.4689	104.9377	81.0453
17.0000	0.3450	1.0000	79.3447	-52.4689	104.9378	81.0454
(其余的数据省略)						
20.0000	0.3550	1.0000	78.2003	-52.4683	104.9366	81.0442
20.0000	0.3600	1.0000	78.1201	-52.4684	104.9368	81.0444
20.0000	0.3650	1.0000	78.0409	-52.4685	104.9370	81.0446
20.0000	0.3700	1.0000	77.9628	-52.4687	104.9373	81.0449
20.0000	0.3750	1.0000	77.8857	-52.4689	104.9377	81.0453
20.0000	0.3800	1.0000	77.8096	-52.4691	104.9382	81.0458
20.0000	0.3850	1.0000	77.7345	-52.4693	104.9387	81.0463
20.0000	0.3900	1.0000	77.6603	-52.4696	104.9392	81.0468
20.0000	0.3950	1.0000	77.5871	-52.4699	104.9399	81.0475
20.0000	0.4000	1.0000	77.5148	-52.4703	104.9406	81.0482

REML Estimation Iteration History

Iteration	Evaluations	Objective	Criterion	
1	2	81.04176181	0.00000000	只经过一次的循环估计就达收敛的指标。

Convergence criteria met.

Covariance Parameter Estimates (REML)

Cov Parm	Ratio	Estimate	Std Error	Z	Pr > Z	
B	18.57323367	1464.3619789	2098.0085361	0.70	0.4852	共变异数的估计值
A*B	0.34192333	26.95812296	59.65698370	0.45	0.6514	B 的标准误差显然比
Residual	1.00000000	78.84259709	35.35117395	2.23	0.0257	其他两个效果更大。

Asymptotic Covariance Matrix of Estimates

Cov Parm	B	A*B	Residual
B	4401639.8175	1.28306865	-273.3162447
A*B	1.28306865	3558.9557045	-502.8436514
Residual	-273.3162447	-502.8436514	1249.7054996

Description	Value
Observations	16.0000
Variance Estimate	78.8426
Standard Deviation Estimate	8.8793
REML Log Likelihood	-52.4671
Akaike's Information Criterion	-55.4671
Schwartz's Bayesian Criterion	-56.3145
-2 REML Log Likelihood	104.9342
PARMS Model LRT Chi-Square	0.0000
PARMS Model LRT DF	2.0000
PARMS Model LRT P-Value	1.0000

卡平方检定不显著，因为最优的
模型与现有的模型十分近似。

Mixed Model Equations (等于一般线性模型中的正规方程序)

Parameter	Row	COL1	COL2	COL3	COL4	
INTERCEPT	1	0.20293598	0.06341749	0.07610099	0.06341749	
A 1	2	0.06341749	0.06341749	0.00000000	0.00000000	固定效果
A 2	3	0.07610099	0.00000000	0.07610099	0.00000000	
A 3	4	0.06341749	0.00000000	0.00000000	0.06341749	
B 1	5	0.43729330	0.16398499	0.16398499	0.10932333	
B 2	6	0.43729330	0.10932333	0.16398499	0.16398499	
A*B 1 1	7	0.02224973	0.02224973	0.00000000	0.00000000	
A*B 1 2	8	0.01483315	0.01483315	0.00000000	0.00000000	随机效果
A*B 2 1	9	0.02224973	0.00000000	0.02224973	0.00000000	
A*B 2 2	10	0.02224973	0.00000000	0.02224973	0.00000000	
A*B 3 1	11	0.01483315	0.00000000	0.00000000	0.01483315	
A*B 3 2	12	0.02224973	0.00000000	0.00000000	0.02224973	
Y	13	36.41432558	13.87574789	12.74691648	9.79166121	因变量 Y
		COL5	COL6	COL7	COL8	COL9
		0.43729330	0.43729330	0.02224973	0.01483315	0.02224973
		0.16398499	0.10932333	0.02224973	0.01483315	0.00000000
		0.16398499	0.16398499	0.00000000	0.00000000	0.02224973
		0.10932333	0.16398499	0.00000000	0.00000000	0.00000000
		1.89727222	0.00000000	0.09588893	0.00000000	0.09588893
		0.00000000	1.89727222	0.00000000	0.06392595	0.00000000
		0.09588893	0.00000000	0.02569385	0.00000000	0.00000000
		0.00000000	0.06392595	0.00000000	0.02135707	0.00000000
		0.09588893	0.00000000	0.00000000	0.00000000	0.02569385
		0.00000000	0.09588893	0.00000000	0.00000000	0.00000000
		0.06392595	0.00000000	0.00000000	0.00000000	0.00000000
		0.00000000	0.09588893	0.00000000	0.00000000	0.00000000

91.77693205 65.15670221 5.46601640 2.64771758 4.24969796

COL10	COL11	COL12	COL13
0.02224973	0.01483315	0.02224973	36.41432558
0.00000000	0.00000000	0.00000000	13.87574789
0.02224973	0.00000000	0.00000000	12.74691648
0.00000000	0.01483315	0.02224973	9.79166121
0.00000000	0.06392595	0.00000000	91.77693205
0.09588893	0.00000000	0.09588893	65.15670221
0.00000000	0.00000000	0.00000000	5.46601640
0.00000000	0.00000000	0.00000000	2.64771758
0.00000000	0.00000000	0.00000000	4.24969796
0.02569385	0.00000000	0.00000000	3.20396077
0.00000000	0.02135707	0.00000000	2.73671649
0.00000000	0.00000000	0.02569385	2.98888007
3.20396077	2.73671649	2.98888007	6841.9232738

Parameter	Row	COL1	COL2	COL3	COL4
INTERCEPT	1	761.83876187	-29.77182106	-29.65777242	0.00000000
A 1	2	-29.77182106	59.54364213	29.77182106	0.00000000
A 2	3	-29.65777242	29.77182106	56.27726675	0.00000000
A 3	4	0.00000000	0.00000000	0.00000000	0.00000000
B 1	5	-731.1427956	-2.07638780	-1.03819390	0.00000000
B 2	6	-733.2191834	2.07638780	1.03819390	0.00000000
A*B 1 1	7	-0.46795069	-14.02387450	0.46795069	0.00000000
A*B 1 2	8	0.46795069	-12.93424846	-0.46795069	0.00000000
A*B 2 1	9	-0.52570042	1.05140084	-12.95336106	0.00000000
A*B 2 2	10	0.52570042	-1.05140084	-14.00476190	0.00000000
A*B 3 1	11	-12.46629777	12.93424846	12.46629777	0.00000000
A*B 3 2	12	-14.49182519	14.02387450	14.49182519	0.00000000
Y	13	159.61444114	53.20486763	7.88555886	0.00000000-随机效果的估计值

固定效果的估计值

COL5	COL6	COL7	COL8	COL9
-731.1427956	-733.2191834	-0.46795069	0.46795069	-0.52570042
-2.07638780	2.07638780	-14.02387450	-12.93424846	1.05140084
-1.03819390	1.03819390	0.46795069	-0.46795069	-12.95336106
0.00000000	0.00000000	0.00000000	0.00000000	0.00000000
741.63175148	722.73022743	-4.25979252	4.25979252	-4.78549295
722.73022743	741.63175148	4.25979252	-4.25979252	4.78549295
-4.25979252	4.25979252	22.80267255	4.15545042	2.15699084
4.25979252	-4.25979252	4.15545042	22.80267255	-2.15699084

-4.78549295	4.78549295	2.15699084	-2.15699084	22.55604292
4.78549295	-4.78549295	-2.15699084	2.15699084	4.40208004
-4.25979252	4.25979252	1.92003907	-1.92003907	2.15699084
4.25979252	-4.25979252	-1.92003907	1.92003907	-2.15699084
26.88369014	-26.88369014	3.01976602	-3.01976602	-1.71336716

COL10	COL11	COL12	COL13
0.52570042	-12.46629777	-14.49182519	159.61444114
-1.05140084	12.93424846	14.02387450	53.20486763
-14.00476190	12.46629777	14.49182519	7.88555886
0.00000000	0.00000000	0.00000000	0.00000000
4.78549295	-4.25979252	4.25979252	26.88369014
-4.78549295	4.25979252	-4.25979252	-26.88369014
-2.15699084	1.92003907	-1.92003907	3.01976602
2.15699084	-1.92003907	1.92003907	-3.01976602
4.40208004	2.15699084	-2.15699084	-1.71336716
22.55604292	-2.15699084	2.15699084	1.71336716
-2.15699084	22.80267255	4.15545042	-0.81148446
2.15699084	4.15545042	22.80267255	0.81148446
1.71336716	-0.81148446	0.81148446	13.00000000

等于样本数减去 X 矩阵的秩。

Tests of Fixed Effects

Source	NDF	DDF	Type III F	Pr > F
A	2	2	28.00	0.0345

Least Squares Means

Level	LSMEAN	Std Error	DDF	T	Pr > T
A 1	212.81930877	27.60142681	10	7.71	0.0000
A 2	167.50000000	27.54633340	10	6.08	0.0001
A 3	159.61444114	27.60142681	10	5.78	0.0002


A 效果的显著来自
第一组与其他两组
之不同。

Predicted Values

obs_No	Observed	Predicted	Var Pred	SE Pred	L95M	U95M	Residual
1	237.0000	242.7228	22.3316	4.7256	232.1934	253.2521	-5.7228
2	254.0000	242.7228	22.3316	4.7256	232.1934	253.2521	11.2772
3	246.0000	242.7228	22.3316	4.7256	232.1934	253.2521	3.2772
4	178.0000	182.9159	30.5354	5.5259	170.6034	195.2283	-4.9159
5	179.0000	182.9159	30.5354	5.5259	170.6034	195.2283	-3.9159
6	208.0000	192.6703	22.0972	4.7008	182.1964	203.1443	15.3297

7	178.0000	192.6703	22.0972	4.7008	182.1964	203.1443	-14.6703
8	187.0000	192.6703	22.0972	4.7008	182.1964	203.1443	-5.6703
9	146.0000	142.3297	22.0972	4.7008	131.8557	152.8036	3.6703
10	145.0000	142.3297	22.0972	4.7008	131.8557	152.8036	2.6703
11	141.0000	142.3297	22.0972	4.7008	131.8557	152.8036	-1.3297
12	186.0000	185.6866	30.5354	5.5259	173.3742	197.9991	0.3134
13	183.0000	185.6866	30.5354	5.5259	173.3742	197.9991	-2.6866
14	142.0000	133.5422	22.3316	4.7256	123.0129	144.0716	8.4578
15	125.0000	133.5422	22.3316	4.7256	123.0129	144.0716	-8.5422
16	136.0000	133.5422	22.3316	4.7256	123.0129	144.0716	2.4578

上述分析的结果显示：各观察体的 Y 值与其预测的值十分接近，若想进一步评估误差的大小，读者可用 G3D 程序将预测之可能率值针对 COL1 (B 效果) 及 COL2 (A*B 效果) 做图：



28.6 注意事项

■ 遗漏数据的处理

MIXED 程序在估计参数的值时，只考虑完全没有遗漏数据的观察体。然而，若观察体只在因变量 Y 上无数据，则 MIXED 程序仍可计算该观察体的预测值。若观察体的遗漏数据是自变量值，组别值或任何其他资料，则 MIXED 程序无法计算其预测值。

■ MIXED 程序与其他 SAS 程序的比较

MIXED 程序是 GLM 程序(第 31 章)的推广：GLM 程序只接受一般线性模型，MIXED 程序接受各式的模型，其中含一般线性模型。两个程序的语法都含 CLASS, MODEL, CONTRAST, ESTIMATE, LSMEANS 等指令，然而其 RANDOM 与 REPEATED 指令的含意迥异。MIXED 程序中的 RANDOM 指令将随机效果视为混合式模型中的 v 向量，然而 GLM 程序视随机效果如同固定效果，因此 GLM 程序对随机效果的估计法与对固定效果的估计法完全一样。但是在 MIXED 程序里，随机效果的变异量是以 REML 或 ML 方法来估计的，与 GLM 程序用的 ANOVA 不同。根据 Searle 等人 (1988, 1992)

的看法，REML 或 ML 法优于 ANOVA 法。

同理，REPEATED 指令在 MIXED 和 GLM 程序中的意义也不相同。MIXED 程序中 REPEATED 指令是用来处理重复观察值的变异数结构的推测，GLM 程序中的 REPEATED 指令是用来转换两个或两个以上的因变量值以便进行单变量或多变量的变异数分析。在处理遗漏数据方面，MIXED 程序较 GLM 程序占优势。

若将 MIXED 与第 27 章的 VARCOMP 程序作比较，则读者不难发现：前者所能处理的线性随机模型远超过 VARCOMP，而且分析的过程更详尽也更周全。不过，VARCOMP 的报表含四种估计变异数成分的方法（即 TYPE I，II，III 与 IV），MIXED 程序只提供第三类型的变异数成分。

第 29 章 变异数分析的实验设计：统计程序 PROC PLAN

29.1 PROC PLAN 程序的简介

PLAN 程序的主要功能在于制造实验中随机取样的各种可能性，以便读者从其中自选所需的样本设计。在任何一种实验设计里，一个细格的识别代号可用其所对应的主效果的组别来表示。比方说，二元的实验设计含 A, B 两个主效果。A 效果下分三组：即 A1 (电脑教学组), A2 (传统式的授课方式)、A3 (学生分组讨论式)；B 效果下则分两组：B1 (重点高中)、B2 (非重点高中)。因此，所有非重点高中里采用电脑教学的班级都归入 (1, 2) 的细格。

当读者选用 PLAN 程序来进行随机地抽样时，它首先会针对第一个主效果 (亦即上例中的 A 效果) 随机抽取所需的组数。然后，根据 A 效果的组别，再进行第二效果 (即上例中的 B 效果) 的随机取样。

取样的方式共三种，随机取样只是其中的一种而已。有关这三种方式的详细介绍，请见第 29.2 节指令 #2 FACTORS 的介绍。

此外，PLAN 程序也容许读者以 TREATMENTS 指令界定一个 (或多个) 的实验效果。这个 (或这些) 实验效果的取样过程与 FACTORS 指令中最后一个界定的主效果同时进行，请看下例：

```
PROC PLAN;
    FACTORS REP=3 ORDERED DRUG=4;
    TREATMENTS DOSE=4;
RUN;
```

这个程序的用意是产生一个 3×4 的三元实验设计；在此设计中，DOSE 的 4 种份量 (以组别 1, 2, 3, 4 为代表) 是按随机的方式分配到每一细格内的。这个程序的报表如下所示 (指令 TREATMENTS 的详细介绍请参阅第 29.2 节的指令 #4)：

The SAS System			
Procedure PLAN			
Plot Factors			
Factor	Select	Levels	Order
REP	3	3	Ordered
DRUG	4	4	Random
Treatment Factors			
Factor	Select	Levels	Order
DOSE	4	4	Random

REP [DRUG DOSE]				
-----+-----+-----+				
1	[2 4]	[3 3]	[4 1]	[1 2]
2	[4 4]	[2 1]	[3 3]	[1 2]
3	[3 3]	[1 1]	[2 2]	[4 4]

29.2 如何撰写 PROC PLAN 程序

PROC PLAN 含四道指令，它们的格式如下：

PROC PLAN	选项串；
FACTORS	主效果的抽样方式 / NOPRINT；
OUTPUT	OUT=输出文件名称 DATA=SAS 文件名称 主效果的抽样结果；
TREATMENTS	其它效果的抽样方式；

其中，PROC PLAN 与 FACTORS 两指令是必须的。

指令 #1 PROC PLAN 选项串；

有两个选项：SEED= 以及 ORDERED，分述如下：

(1) SEED=一个小于 $(2^{31}-1)$ 的整数，如 54321

此选项的作用在于启动一个随机的排列。若省略此选项，则 PLAN 程序利用电脑内的电钟时间作为启动的数字。

(2) ORDERED

要求将主效果的组别以整数顺序 1, 2, ..., m 等来代表。此选项与下一个指令 FACTORS 联用，请参见下一段的范例。

注：程序中也可界定一个 DATA=SAS 文件的语句，不过这一个语句是放在指令 #3 OUTPUT 里，而非此处 (即指令 #1)。

指令 #2 FACTORS 主效果的抽样方式 / NOPRINT；

此指令分两部分：主效果的抽样方式以及选项 NOPRINT。现分别说明如下页：

主效果的抽样方式

格式如下：

主效果名称=m [OF n] 抽样方式；

在此，主效果名称= 实验设计中的主效果代号，也是一个合于 SAS 语法的变量名称。

m = 小于或等于 n 的正整数，代表主效果在样本中的组别数目。

n = 正整数，代表主效果在母群中的组别数目。

抽样方式=RANDOM(内设值) 或 ORDERED 或 CYCLIC。

一个 FACTORS 指令中可含数个主效果的抽样方式，它们之间必须以一个空白隔开；中括号[] 的部分可有可无，m 和 n 都必须为正整数，而且 $m \leq n$ 。三种抽样方式定义如下：

RANDOM = 自 n 组中, 随机抽取 m 个组; 或将 m 组按随机方式排列。

ORDERED = 将 m 组界定为 1 到 m 的整数, 因此, 这种方法一定不会牵涉到随机抽样。

CYCLIC = 循环式地界定组别。有三种撰写方法, 请参见下表中的例三~例五。

现举几例说明这些抽样方式的语法以及抽样的结果:

表 29.1 实验设计中抽样方式的语法示

	语法的形式	范例	结果
例一	效果= m	A=7	将 1 到 7 的整数随机排列。
例二	效果= m ORDERED	B=3 ORDERED	将 1 到 3 的整数由小至大排列。
例三	效果= m CYCLIC	C=5 CYCLIC	第一回合, 排列 (1, ..., 5); 第二回合, 排列 (2, 3, 4, 5, 1); 以此类推。
例四	效果= m CYCLIC i	D=4 CYCLIC 2	第一回合, 排列 (1, ..., 4); 第二回合, 排列 (3, 4, 1, 2); 第三回合, 排列 (1, 2, 3, 4)。
例五	效果= m OF n CYCLIC (m 个整数) i	E=4 OF 10 CYCLIC (1 3 7 9) 2	第一回合的组别排列是 (1, 3, 7, 9); 第二回合的组别排列是 (3, 5, 9, 1)。
例六	效果= m OF n	F=3 OF 7	自 1 到 7 的整数中, 随机取出 3 个整数。
例七	效果= m OF n ORDERED	G=2 OF 4 ORDERED	将 1, 2 按顺序排; 也等于 G=2 ORDERED。
例八	效果 1=M1 效果 2=M2	I=4 J=3	效果 1 的四组与效果 J 的三组都会按随机的 方式排列; 而且, J 是镶嵌在 I 效果内。
例九	效果 1=M1 ORDERED 效果 2=M2	K=2 ORDERED L=3	将 (1, 2, 3) 三个数字随机排列两次。

删除号 (/) 后面只有一个选项, 介绍如下:

(1) NOPRINT

抑止任何报表的打印。若读者只想将实验设计的结果存在一个输出文件内, 则此选项特别有用。

指令 #3 OUTPUT OUT=输出文件名称 DATA=SAS 文件名称 主效果的抽样结果:

这个指令应与一个已知的实验设计联用; 因此, 当读者每次界定此指令之前, 必须首先界定一个 FACTORS 或 TREATMENTS 的指令。

指令 #3 分三部分, 即 OUT=, DATA= 以及「主效果的抽样结果」。为什么 OUTPUT 指令容许读者同时引用一个输出文件以及另一个 (输入) 文件呢? 这是因为 PLAN 程序所产生的实验设计可转变成合法的 SAS 输出文件 (第一种情况, 只需界定 OUT= 即可)。或者, PROC PLAN 的结果可用来改变另一个 SAS 文件的实验设计 (第二种情况, 必须同时界定 OUT= 与 DATA=两关键字)。这两种情况会直接影响「主效果的抽样结果」的撰写方式, 因此, 下面仅就这两种情况分别介绍 OUTPUT 指令的语法:

第一种情况: 将实验的设计输出

在这种情况下, 读者只需界定 OUT= 输出文件名称以及「主效果的抽样结果」。OUT= 输出资料内含实验设计里所提到的每一主效果 (即变量名称) 以及主效果的组别 (即变量的值)。观察体的数目则等于实验设计里细格的总数。

主效果的抽样结果

其语法不外乎下列两种之一:

主效果名称 NVALS=(含 n 个数字的数字串) ORDERED (或 RANDOM)

[内设值] NVALS=(1 2 ... n)

或主效果名称 CVALS=(含 n 个标签的文字串) ORDERED (或 RANDOM)

请看下页的举例说明：

```
PROC PLAN;
  FACTORS A=6 ORDERED B=3;
  OUTPUT OUT=EXP
    A NVALS=(10 20 30 40 50 60)
    B CVALS=('DAY1' 'DAY2' 'DAY3');
RUN;
```

由此例的示范我们应注意到括号内数字串或文字串的个数应与 A, B 的组数相等, 因为这些字串就是各组的命名。此外, 文字串必须一一以单引号括住, 其长度不超过 40 个字母。

最后解释 ORDERED (或 RANDOM) 的意义：这两个关键字的用意是指示 PLAN 程序如何将上述 NVALS= 或 CVALS= 所界定的字串与主效果的第一、第二、...、第 n 组配对。若上例的 OUTPUT 部分润饰如下：

```
OUTPUT OUT=EXP
  A NVALS=(10 20 30 40 50 60) ORDERED
  B CVALS=('DAY1' 'DAY2' 'DAY3') RANDOM;
```

则 A 效果的第一组就是 A=10 的那一组, 第二组就是 A=20 的组, 以此类推。然而, B 效果的第一, 二, 三组则是以随机配对的方式与 'DAY1', 'DAY2', 'DAY3' 等组名相联接。

根据上面的解释, 读者当可理解：在这两个关键字中 ORDERED 是内设值。

第二种情况：改变另一个 SAS 文件的实验设计

在这种情况下, 读者必须同时界定 OUT= 输出文件名称以及 DATA=SAS 文件名称。此外, PLAN 程序假设每一个主效果名称 (属 OUT= 文件) 也就是 DATA= 文件的一个变量名称。若此假设不成立, 则读者可用下列的等式将两个互相对应的名字串连在一起：

变量名称 (属 DATA=文件)=主效果名称 (属 OUT= 文件)

接下来, 读者可参照上节第一种情况的说明来界定主效果(或 DATA= 文件内的另一个变量)的抽样结果。有关第二种情况的语法示范, 请参考第 29.3 节的例一。

指令 #4 TREATMENTS 其他效果的抽样方式：

这个指令所界定的其他效果必定是镶嵌在实验设计的每一细格内的。因此, 这个指令所提及的 "其它效果" 并不是实验设计的主效果；故 TREATMENTS 与 FACTORS 两指令的功用不同。

对 PLAN 程序而言, 无论读者界定的 TREATMENTS 指令有多少个, 只有最后一

个 (也就是最靠近 RUN; 指令的那一个) 才是有效的。

这个指令的撰写方式与指令 #2 FACTORS 完全相同。唯一值得读者注意的是：这个指令不附加删除号 (/) 或 NOPRINT 的选项。然而，FACTORS 指令之后可以附加这两项。

其它效果的抽样方式

其语法如下：

其它效果名称 = m [OF n] 抽样方式；

有关 m, n 的定义以及抽样方式的代号，如 RANDOM (内设值)、ORDERED，或 CYCLIC 等，请回头参考指令 #2 FACTORS 主效果的抽样方式之说明 (表 29.1)，在此不另赘述。

除了第 29.1 节所举的范例之外，下面再举一例说明指令 #2, #4 的写法：

```
PROC PLAN;  
  FACTORS FIRST=4 ORDERED SECOND=4 ORDERED;  
  TREATMENTS A=4 CYCLIC  
              B=4 CYCLIC 2;  
  
RUN;
```

上面的程序所导出的实验设计是一个 4×4 的 Graeco 拉丁方格的设计，其主效果各是 FIRST 与 SECOND；两个中介效果各是 A, B。每一效果的组别数都是 4。

执行上述程序后，你可能得到如下的报表结果：

The SAS System					
Procedure PLAN					
Plot Factors					
<u>Factor</u>	<u>Select</u>	<u>Levels</u>	<u>Order</u>		
FIRST	4	4	Ordered		
SECOND	4	4	Ordered		
Treatment Factors					
<u>Factor</u>	<u>Select</u>	<u>Level</u>	<u>Order</u>	<u>Initial block / Increment</u>	
		<u>s</u>			
A	4	4	Cyclic	(1 2 3 4) / 1	
B	4	4	Cyclic	(1 2 3 4) / 2	
FIRST [SECOND A B]					
-----	-----	-----	-----	-----	-----
1	[1 1 1]	[2 2 2]	[3 3 3]	[4 4 4]	
2	[1 2 3]	[2 3 4]	[3 4 1]	[4 1 2]	
3	[1 3 1]	[2 4 2]	[3 1 3]	[4 2 4]	
4	[1 4 3]	[2 1 4]	[3 2 1]	[4 3 2]	

根据报表的设计，我们很容易看出：FIRST 与 SECOND 两主效果的组别排列是按顺序，由小到大。然而，A 与 B 的排列则以轮回的方式排列。

COMPLETELY RANDOMIZED DESIGN

OBS	UNIT	TREAT
1	1	1
2	2	1
3	3	2
4	4	1
5	5	1
6	6	1
7	7	2
8	8	1
9	9	2
10	10	2
11	11	2
12	12	2

例二：重复观察的实验设计

在这个重复观察的实验里，假设有三个主要区间（或称区集，英文是 BLOCK）。每个区间内分成四个畦区（Plot），每个畦区内再细分成两个较小的次畦区（Subplot）。

我们以 BLOCK 表区间，A 表畦区，B 表次畦区，下列的程序可帮助我们决定：在每一个区间内，畦区被分派在各实验组的随机安排。同时，在一畦区内，我们也可预先设计次畦区的随机安排。

程 序

```
PROC PLAN SEED=3;  
  TITLE 'SPLIT PLOT DESIGN';  
  FACTORS BLOCK=3 ORDERED A=4 B=2;  
  
RUN;
```

结 果

报表 29.2 重复观察的实验设计

SPLIT PLOT DESIGN

Procedure PLAN

Factor	Select	Levels	Order
BLOCK	3	3	Ordered
A	4	4	Random
B	2	2	Random

BLOCK	A	B
1	3 4 2 1	2 1 2 1 2 1 1 2

2	3	1	2
	4	2	1
	2	2	1
	1	2	1
3	2	2	1
	4	2	1
	3	2	1
	1	2	1

例三：层次式实验设计 (Hierarchical Design)

这个例子含三个自变量，即：房子 (HOUSES)、盆栽 (POTS) 与 植物 (PLANTS)。它们各自的组数是 3, 4 与 3。下列的 PLAN 程序会帮读者作几件事：随机排列三栋房子的代号；在每一个房子代号下随机排列四个盆栽的代号，于是产生十二个组合。最后，再在每一栋房子和盆栽的组合下，随机排列三株植物的代号。由于植物镶嵌在盆栽里，盆栽镶嵌在房子里，因此，这实验设计属于层次式实验设计。

程 序

```
PROC PLAN SEED=17431;
  TITLE 'HIERARCHICAL DESIGN';
  FACTORS HOUSES=3 POTS=4 PLANTS=3;
RUN;
```

结 果

报表 29.3 层次式实验设计

HIERARCHICAL DESIGN				
Procedure PLAN				
Factor	Select	Levels	Order	
-----	-----	-----	-----	
HOUSES	3	3	Random	
POTS	4	4	Random	
PLANTS	3	3	Random	
HOUSES	POTS	PLANTS		
-----	-----	---+---+---+		
1	3	3	1	2
	1	1	2	3
	2	2	1	3
	4	3	1	2
3	4	2	1	3
	1	3	1	2
	3	1	3	2
	2	1	3	2
2	1	2	1	3
	2	3	2	1
	4	1	2	3
	3	1	3	2

例四：通用畦区设计

本例示范如何撰写 PLAN 程序的指令以便产生一个通用的畦区实验设计。实验本身含八组，是从母群的五十二组中以随机方式抽选出来的；每一组内又含十三个畦区 (BLOCKS)，并且每一畦区内含八个次畦区 (PLOTS)。因此，BLOCKS=13，PLOTS=8。

由 PLAN 程序产生的实验设计再输入 TABULATE 程序以便检视实验设计是否完全根据随机抽样的原则。

程 序

```
TITLE 'GENERALIZED CYCLIC BLOCK DESIGN';
PROC PLAN SEED=33373;
  FACTORS BLOCKS=13 PLOTS=8;
  TREATMENTS TRTMTS=8 OF 52 CYCLIC (1 2 3 4 32 43 46 49) 4;
  OUTPUT OUT=C;
  QUIT;
PROC TABULATE;
  CLASS BLOCKS PLOTS;
  VAR TRTMTS;
  TABLE BLOCKS, PLOTS*(TRTMTS*F=8.)/RTS=8;
RUN;
```

结 果

报表 29.4 通用畦区设计

GENERALIZED CYCLIC BLOCK DESIGN									
Procedure PLAN									
Plot Factors									
Factor	Select	Levels	Order						
BLOCKS	13	13	Random						
PLOTS	8	8	Random						
Treatment Factors									
Factor	Select	Levels	Order	Initial block / Increment					
TRTMTS	8	52	Cyclic	(1 2 3 4 32 43 46 49) / 4					
BLOCKS [PLOTS TRTMTS]									
10	[7 1]	[4 2]	[8 3]	[1 4]	[2 32]	[3 43]	[5 46]	[6 49]	
8	[1 5]	[2 6]	[4 7]	[3 8]	[8 36]	[6 47]	[5 50]	[7 1]	
9	[2 9]	[5 10]	[4 11]	[7 12]	[3 40]	[1 51]	[8 2]	[6 5]	
6	[4 13]	[2 14]	[6 15]	[8 16]	[3 44]	[7 3]	[1 6]	[5 9]	
7	[4 17]	[7 18]	[6 19]	[3 20]	[1 48]	[2 7]	[8 10]	[5 13]	
4	[4 21]	[8 22]	[1 23]	[5 24]	[3 52]	[6 11]	[7 14]	[2 17]	
2	[6 25]	[2 26]	[3 27]	[8 28]	[7 4]	[5 15]	[1 18]	[4 21]	
3	[6 29]	[2 30]	[3 31]	[1 32]	[7 8]	[4 19]	[5 22]	[8 25]	
1	[1 33]	[2 34]	[7 35]	[8 36]	[5 12]	[6 23]	[3 26]	[4 29]	
5	[5 37]	[7 38]	[6 39]	[8 40]	[4 16]	[3 27]	[1 30]	[2 33]	

```

12 [ 5 41] [ 8 42] [ 1 43] [ 4 44] [ 7 20] [ 3 31] [ 6 34] [ 2 37]
13 [ 3 45] [ 5 46] [ 1 47] [ 8 48] [ 4 24] [ 2 35] [ 6 38] [ 7 41]
11 [ 4 49] [ 1 50] [ 5 51] [ 2 52] [ 3 28] [ 8 39] [ 6 42] [ 7 45]

```

下面是 PROC TABULATE 的结果

	PLOTS							
	1	2	3	4	5	6	7	8
	TRTMTS	TRTMTS	TRTMTS	TRTMTS	TRTMTS	TRTMTS	TRTMTS	TRTMTS
	SUM	SUM	SUM	SUM	SUM	SUM	SUM	SUM
BLOCKS								
1	33	34	26	29	12	23	35	36
2	18	26	27	21	15	25	4	28
3	32	30	31	19	22	29	8	25
4	23	17	52	21	24	11	14	22
5	30	33	27	16	37	39	38	40
6	6	14	44	13	9	15	3	16
7	48	7	20	17	13	19	18	10
8	5	6	8	7	50	47	1	36
9	51	9	40	11	10	5	12	2
10	4	32	43	2	46	49	1	3
11	50	52	28	49	51	42	45	39
12	43	37	31	44	41	34	20	42
13	47	35	45	24	46	38	41	48

例五：拉丁方格的实验设计

本例示范如何产生一个 4×4 的拉丁方格的设计。程序中 ROWS 与 COLS 均代表中介变量，TMTS 则代表实验效果，必须与每一细格相对应 (因此，由 TREATMENTS 指令来界定)。ROWS 变量下的组别分别是 'Day 1', 'Day 2', 'Day 3' 及 'Day 4'。COLS 变量下的组别分别是 'Lab 1', 'Lab 2', 'Lab 3', 'Lab 4' 等。TMTS 的四组则是以 0, 100, 250, 450 等数值来代表。

如同例四的程序，PLAN 程序所产生的实验设计再经 TABULATE 程序处理后打印以便确定是一个根据随机原理设计的 4×4 之拉丁方格实验。

程 序

```

TITLE 'LATIN SQUARE DESIGN';
PROC PLAN SEED=37430;
    FACTORS ROWS=4 ORDERED COLS=4 ORDERED/NOPRINT; [PROC PLAN 的结果不会在报表上显出来]

    OUTPUT OUT=G
        ROWS CVALS=('Day 1' 'Day 2' 'Day 3' 'Day 4') RANDOM
        COLS CVALS=('Lab 1' 'Lab 2' 'Lab 3' 'Lab 4') RANDOM
        TMTS NVALS=( 0      100      250      450 ) RANDOM;
    TREATMENTS TMTS=4 CYCLIC;
    QUIT;
PROC TABULATE;
    CLASS ROWS COLS;
    VAR TMTS;

```

```
TABLE ROWS, COLS*(TMTS*F=6.) /RTS=8;  
RUN;
```

结 果

报表 29.5 拉丁方格的实验设计

LATIN SQUARE DESIGN				
	COLS			
	Lab 1	Lab 2	Lab 3	Lab 4
	TMTS	TMTS	TMTS	TMTS
	SUM	SUM	SUM	SUM
	ROWS			
Day 1	0	100	450	250
Day 2	100	250	0	450
Day 3	250	450	100	0
Day 4	450	0	250	100

29.4 注 意 事 项

■在交谈式环境下执行 PLAN 程序

所有 PLAN 程序的指令都适用于交谈式环境里。然而，值得读者注意的是，若你想结束 PLAN 程序 (如上节例四与例五的程序) 以便执行另一个统计程序，如 TABULATE，则必须附加上 QUIT；的指令 (而非 RUN；的指令)。

若将 RUN；指令附加在 PLAN 程序之后，则它的作用是启动一个设计的过程 (或执行 PLAN 程序的设定)，而非结束。因此，RUN；指令之后可添加其他的设计语句如 FACTORS, TREATMENTS, OUTPUT 等，但不可引用另一个以 PROC 起头的统计程序。

第 30 章 无参数的一因子变异数分析：统计程序 PROC NPAR1WAY

30.1 PROC NPAR1WAY 程序概述

NPARIWAY 程序对数据的名次 (而非数据本身), 执行一因子变异数分析。其基本假设是：各组之间的平均名次无显著差异。用 NPAR1WAY 程序分析出来的结果与读者使用 PROC RANK, 再用 PROC ANOVA 分析的结果是完全一致的。在分析的过程中, NPAR1WAY 程序会同时导出四种数值, 分别解释如下：

- (1) Wilcoxon 值
亦即数据的名次, 最小的数据取名次 1, 第二小的数据取名次 2, 以此类推。此值最适用于逻辑斯谛 (Logistic) 的分配。
- (2) Median 值
比中数大的数据, 其 Median 值为 1, 否则等于 0。此值最适用于双指数分配 (DoubleExponential Distribution)。
- (3) van der Waerden 值
是标准常态分配上标准值 (Z_i) 的近似值。此值最适用于常态分配。
- (4) Savage 值
是指数分配上排序的期待值 (也是平均数) 减 1。减 1 的目的是促使平均数的值固定在 0 上。此值最适合用来比较由指数分配所导出的平均数。

下页的表格列出这四种数值与时下一统计般教科书中所提到的无参数分析法的对应：

表 30.1 PROC NPAR1WAY 程序中四种无参数的平均数比较值

NPARIWAY 程序	传统的无参数分析	
	两组数据	三组或三组以上
(1) Wilcoxon 值	Wilcoxon rank sum 检验值 或 Mann-Whitney U 检验值	Kruskal-Wallis 检验值
(2) Median 值	中数检验值	多样本中数检验值(Brown-Mood 值)
(3) van der Waerden 值	van der Waerden 检验值	多样本的 van der Waerden 检验值
(4) Savage 值	Savage 检验值	多样本的 Savagen 检验值

30.2 如何撰写 PROC NPAR1WAY 程序

PROC NPAR1WAY 含五道指令, 它们的格式如下：

```

PROC NPARIWAY 选项串;
    VAR          因变量名称串;
    CLASS        变量名称;
    EXACT;
    BY           变量名称串;

```

其中 PROC NPARIWAY 与 CLASS 两道指令是必需的，不可省略。

指令 #1 PROC NPARIWAY 选项串:

有下列八个选项：

(1) DATA=输入文件名称

指明到底对那一个 SAS 文件执行 NPARIWAY 分析。若省略此选项，则 SAS 会自动找出在此程序之前最后形成的 SAS 文件，对它执行分析。

(2) MISSING

将自变量上有遗漏数据的观察体归成一组，且将此组纳入分析中。

若读者没有选用下列任何六选项之一，则 NPARIWAY 程序自动将它们全部包括进去：

(3) ANOVA

要求 SAS 除了无参数的变异数分析外，另外也执行传统的变异数分析。

(4) WILCOXON

规定 SAS 将原数据转换成从小到大的名次，然后进行无参数的变异数分析。这个分析的结果是一般的 Wilcoxon Rank Sum 检定 (若只有两组数据)，或是 Kruskal-Wallis 检定 (若含三组或以上的数据)。

(5) MEDIAN

规定 SAS 将原数据二分：凡是大于中数的数据都转换成 1，否则转换成 0。然后进行无参数的变异数分析。这个分析的结果是一般的中数检定 (若只有两组数据)，或 Brown-Mood 测试 (若含三组或以上的数据)。

(6) VW

要求 SAS 计算出 van der Waerden 的值 (简称 VW 值)，然后进行无参数的变异数分析。VW 值依下式导出：

$$Z_i = \Phi^{-1} [R_i / (n+1)]$$

在此， Φ^{-1} 是常态分配的反函数，

R_i 是数据的名次，

n 是样本的大小，

Z_i 是 VW 值。

(7) SAVAGE

规定 SAS 将原数据先转换成 Savage 值，然后进行无参数的变异数分析。转换式如下：

$$Z_i = \sum_{j=1}^{R_i} [1/(n-j+1)] - 1$$

在此, n 是样本的大小,

i 是数据的编号,

R_i 是第 i 个数据的名次 (即 Wilcoxon 值),

Z_i 是第 i 个数据的 Savage 值。

(8) EDF

此选项会根据经验分配函数 (Empirical Distribution Function) 来计算两种至三种的统计值。这些统计值是 Kolmogorov-Smirnov 值、Cramer-von Mises 值, 或 Kuiper 值 (如果组别数等于 2)。

指令 #2 VAR 因变量名称串:

列出因变量的名称。若省略此选项, 则输入文件中所有的数值变量均成为因变量。

指令 #3 CLASS 自变量名称:

由于 NPAR1WAY 程序执行无参数的一因子变异数分析, 所以只能有一个自变量。

指令 #4 EXACT:

此指令用来执行无参数的正准检定 (Exact test)。因此, 针对前述任何一种统计值, EXACT 指令会执行其正准的统计检定。若不选用此指令, 则 NPAR1.WAY 程序执行大样本情况下的卡平方检定。

指令 #5 BY 变量名称串:

NPAR1WAY 程序依据此指令所列举的变量将文件分成几个小的文件, 然后针对每一个小的文件分别执行分析。当读者选用此指令时, 文件内的数据必须先按照 BY 变量串的值做由小到大的重新排列。这个步骤可藉 PROC SORT 达成。

30.3 范 例

例一：体重增加的分析

本文件 (G) 的数据是 Halverson 与 Sherwood 于 1932 年所搜集来的。目的在研讨服用一种药物 (DOSE) 对体重增加 (GAIN) 的影响。

首先, 按照药物服用量的多少将受试者分为五组, 然后对五组数据执行无参数的一因子变异数分析。

第二部分则以服药量等于 .04 毫克为基准, 将数据分成多于或小于 (等于) .04 的两组, 然后比较这两组在体重增减上的差别。

程 序

```

TITLE 'WEIGHT GAINS WITH GOSSYPOL ADDITIVE';
TITLE2 'HALVERSON AND SHERWOOD - 1932';
DATA G;
    INPUT DOSE N;
    DO I=1 TO N;
        INPUT GAIN @;
        OUTPUT;
    END;
CARDS;
0 16
228 229 218 216 224 208 235 229 233 219 224 220 232 200 208 232
.04 11
186 229 220 208 228 198 222 273 216 198 213
.07 12
179 193 183 180 143 204 114 188 178 134 208 196
.10 17
130 87 135 116 118 165 151 59 126 64 78 94 150 160 122 110 178
.13 11
154 130 130 118 118 104 112 134 98 100 104
;
PROC NPAR1WAY;
    CLASS DOSE;
    VAR GAIN;
DATA G2;
    SET G;
    IF DOSE<=.04;
PROC NPAR1WAY;
    CLASS DOSE;
    VAR GAIN;
    TITLE4 'DOSES<=.04';
RUN;

```

结 果

在第一步骤的分析里，五组受试者体重的增减 (GAIN) 均与用药量 (DOSE) 的多少有关 (见反白部分的统计值以及显著程度)。

在第二步骤的分析里，五组被简化成两组。此时，受试者的体重不再与这样的分组有关 (见第二部分报表 DOSE<=.04 下不显著的统计分析值)。由此，可以反推，原先组间的差异是在 .04 毫克至 .13 毫克之间，而非 0 毫克与 .04 (含以上) 毫克之间。

报表 30.1 体重增加的分析

WEIGHT GAINS WITH GOSSYPOL ADDITIVE
HALVERSON AND SHERWOOD - 1932

N P A R 1 W A Y P R O C E D U R E

Analysis of Variance for Variable GAIN
Classified by Variable DOSE

DOSE	N	Mean	Among MS	Within MS
			35020.7465	627.451597
0	16	222.187500		
0.04	11	217.363636	F Value	Prob > F
0.07	12	175.000000	55.814	0.0001
0.1	17	120.176471		
0.13	11	118.363636		

N P A R 1 W A Y P R O C E D U R E

Wilcoxon Scores (Rank Sums) for Variable GAIN
Classified by Variable DOSE

DOSE	N	Sum of Scores	Expected Under H0	Std Dev Under H0	Mean Score
0	16	890.500000	544.0	67.9789655	55.6562500
0.04	11	555.000000	374.0	59.0635883	50.4545455
0.07	12	395.500000	408.0	61.1366221	32.9583333
0.1	17	275.500000	578.0	69.3807412	16.2058824
0.13	11	161.500000	374.0	59.0635883	14.6818182

Average Scores were used for Ties

N P A R 1 W A Y P R O C E D U R E

Kruskal-Wallis Test (Chi-Square Approximation)

CHISQ= 52.666 DF= 4 Prob > CHISQ= 0.0001

Median Scores (Number of Points above Median)
for Variable GAIN
Classified by Variable DOSE

DOSE	N	Sum of Scores	Expected Under H0	Std Dev Under H0	Mean Score
0	16	16.0	7.88059701	1.75790231	1.00000000
0.04	11	11.0	5.41791045	1.52735508	1.00000000
0.07	12	6.0	5.91044776	1.58096271	0.50000000
0.1	17	0.0	8.37313433	1.79415153	0.00000000
0.13	11	0.0	5.41791045	1.52735508	0.00000000

Average Scores were used for Ties

Median 1-Way Analysis (Chi-Square Approximation)

CHISQ= 54.176 DF= 4 Prob > CHISQ= 0.0001

N P A R I W A Y P R O C E D U R E

Van der Waerden Scores (Normal) for Variable GAIN

Classified by Variable DOSE

DOSE	N	Sum of Scores	Expected Under H0	Std Dev Under H0	Mean Score
0	16	16.1164737	0.0	3.32595675	1.00727961
0.04	11	8.3408986	0.0	2.88976066	0.75826351
0.07	12	-0.5766736	0.0	2.99118646	-0.04805613
0.1	17	-14.6889214	0.0	3.39454039	-0.86405420
0.13	11	-9.1917773	0.0	2.88976066	-0.83561612

Average Scores were used for Ties

Van der Waerden 1-Way (Chi-Square Approximation)

CHISQ= 47.297 DF= 4 Prob > CHISQ= 0.0001

Savage Scores (Exponential) for Variable GAIN

Classified by Variable DOSE

DOSE	N	Sum of Scores	Expected Under H0	Std Dev Under H0	Mean Score
0	16	16.0743905	0.0	3.38527520	1.00464941
0.04	11	7.6930992	0.0	2.94129955	0.69937265
0.07	12	-3.5849578	0.0	3.04453428	-0.29874648
0.1	17	-11.9794882	0.0	3.45508203	-0.70467578
0.13	11	-8.2030437	0.0	2.94129955	-0.74573125

Average Scores were used for Ties

Savage 1-Way (Chi-Square Approximation)

CHISQ= 39.491 DF= 4 Prob > CHISQ= 0.0001

Kolmogorov-Smirnov Test for Variable GAIN

Classified by Variable DOSE

DOSE	N	Deviation	
		EDF at maximum	from Mean at maximum
0	16	0.00000000	-1.91044776
0.04	11	0.00000000	-1.58405960
0.07	12	0.33333333	-0.49979576

0.1	17	1.00000000	2.15386115
0.13	11	1.00000000	1.73256519
-----	-----	-----	
	67	0.47761194	

Maximum Deviation occurred at Observation 36
Value of GAIN at maximum 178.000000

Kolmogorov-Smirnov Statistic (Asymptotic)
KS = 0.457928 KSa = 3.74830

Cramer-von Mises Test for Variable GAIN
Classified by Variable DOSE

DOSE	N	Summed Deviation from Mean
0	16	2.16521023
0.04	11	0.91827966
0.07	12	0.34822684
0.1	17	1.49754164
0.13	11	1.33574457

Cramer-von Mises Statistic (Asymptotic)

CM = 0.093508 CMa = 6.26500
DOSES<=.04

WEIGHT GAINS WITH GOSSYPOL ADDITIVE
HALVERSON AND SHERWOOD - 1932
DOSE<=.04

Analysis of Variance for Variable GAIN
Classified by Variable DOSE

DOSE	N	Mean	Among MS	Within MS
			151.683712	271.479318
0	16	222.187500		
0.04	11	217.363636	F Value	Prob > F
			0.559	0.4617

Average Scores were used for Ties
DOSES<=.04

Wilcoxon Scores (Rank Sums) for Variable GAIN
Classified by Variable DOSE

DOSE	N	Sum of Scores	Expected Under H0	Std Dev Under H0	Mean Score
------	---	------------------	----------------------	---------------------	---------------

0	16	253.500000	224.0	20.2215647	15.8437500
0.04	11	124.500000	154.0	20.2215647	11.3181818

Average Scores were used for Ties

Wilcoxon 2-Sample Test (Normal Approximation)

(with Continuity Correction of .5)

S= 124.500 Z= -1.43411 Prob > |Z| = 0.1515

T-Test approx. Significance = 0.1635

Kruskal-Wallis Test (Chi-Square Approximation)

CHISQ= 2.1282 DF= 1 Prob > CHISQ= 0.1446

WEIGHT GAINS WITH GOSSYPOL ADDITIVE

HALVERSON AND SHERWOOD - 1932

DOSES<=.04

Median Scores (Number of Points above Median)

for Variable GAIN

Classified by Variable DOSE

DOSE	N	Sum of Scores	Expected Under H0	Std Dev Under H0	Mean Score
0	16	9.0	7.70370370	1.29999472	0.562500000
0.04	11	4.0	5.29629630	1.29999472	0.363636364

Average Scores were used for Ties

Median 2-Sample Test (Normal Approximation)

S= 4.00000 Z= -.997155 Prob > |Z| = 0.3187

Median 1-Way Analysis (Chi-Square Approximation)

CHISQ= 0.99432 DF= 1 Prob > CHISQ= 0.3187

Van der Waerden Scores (Normal) for Variable GAIN

Classified by Variable DOSE

DOSE	N	Sum of Scores	Expected Under H0	Std Dev Under H0	Mean Score
0	16	3.34651962	0.0	2.32033631	0.209157476
0.04	11	-3.34651962	0.0	2.32033631	-.304229056

Average Scores were used for Ties

Van der Waerden 2-Sample Test (Normal Approximation)

S= -3.34652 Z= -1.44226 Prob > |Z| = 0.1492

Van der Waerden 1-Way (Chi-Square Approximation)

CHISQ= 2.0801 DF= 1 Prob > CHISQ= 0.1492

WEIGHT GAINS WITH GOSSYPOL ADDITIVE

HALVERSON AND SHERWOOD - 1932

DOSES<=.04
Savage Scores (Exponential) for Variable GAIN
Classified by Variable DOSE

DOSE	N	Sum of Scores	Expected Under H0	Std Dev Under H0	Mean Score
0	16	1.83455386	0.0	2.40183886	0.114659616
0.04	11	-1.83455386	0.0	2.40183886	-.166777623

Average Scores were used for Ties

Savage 2-Sample Test (Normal Approximation)

S = -1.83455 Z = -.763812 Prob > |Z| = 0.4450

Savage 1-Way (Chi-Square Approximation)

CHISQ = 0.58341 DF = 1 Prob > CHISQ = 0.4450

Kolmogorov-Smirnov Test for Variable GAIN
Classified by Variable DOSE

DOSE	N	EDF at maximum	Deviation from Mean at maximum
0	16	0.250000000	-.481481481
0.04	11	0.545454545	0.580688515
-----	-----	-----	
	27	0.370370370	

Maximum Deviation occurred at Observation 4

Value of GAIN at maximum 216.000000

Kolmogorov-Smirnov 2-Sample Test (Asymptotic)

KS = 0.145172 D = 0.295455

KSa = 0.754337 Prob > KSa = 0.6199

Cramer-von Mises Test for Variable GAIN
Classified by Variable DOSE

DOSE	N	Summed Deviation from Mean
0	16	0.098638419
0.04	11	0.143474064

Cramer-von Mises Statistic (Asymptotic)

CM = 0.008967 CMa = 0.242112

Kuiper Test for Variable GAIN
Classified by Variable DOSE

		Deviation
DOSE	N	from Mean
0	16	0.090909091
0.04	11	0.295454545
Kuiper 2-Sample Test (Asymptotic)		
K = 0.386364	Ka = 0.986440	Prob > Ka = 0.8383

例二：图形对儿童学习的助益

在这个例题里，我们将示范 EXACT 指令对小样本平均数比较的功能。数据根据十九位维斯康辛州公立学校的二、三年级学生，对两个童话故事的理解程度。研究者首先将这十九位学生随机分成两组：语文组（十位）与图形组（九位）。语文组的学生一边听故事，一边读故事的内容（印在二十张卡片上，由研究者根据口述故事的速度呈现在学生面前）。图形组的学生则一边看故事的图书，一边听口述的故事。每一个故事的细节以图形方式绘于卡片上，由研究者根据口述故事的速度一一呈现在学生前。

受试者听完每一个故事后，即刻回答十题有关该故事内容的问题。答对题数的百分比则代表受试者对故事理解的程度。这个测验的最高分数是 20 分（两个故事，每个故事用十题测之），以变量 SCORE 表之，学生所属的组别以 GROUP 表之。数据收录于附赠之磁片上 DATA 子目录下的 LEARN.DAT 档内。

下面的程式，对这两组学生的分数进行平均数的正确检定以及卡平方等的趋近检定。

程 式

```
OPTIONS NODATE LS=80 PAGENO=1;
TITLE 'EXAMPL 30.2 DO CHILDREN LEARN BETTER FROM PICTURES OR FROM WORDS?';
DATA WIL;
  INFILE 'A:\DATA\LEAEN.DAT';
  INPUT GROUP $ SCORE;
PROC NPARIWAY DATA=WIL WILCOXON;
  CLASS GROUP;
  VAR SCORE;
  EXACT;
RUN;
```

结 果

分析的结果分四部分来解释：

第一部分的报表显示：图形组的平均排名是 15，语文组的平均排名是 5.5。这两组平均名次差是 9.5。这个差异经维尔克森正确检定后，转换成统计值 S=135（第二部分）。这个值经单尾检定后达显著度=1.08E-05（或 0.0000108）。双尾检定的显著度=2.17E-05（也就是 0.0000217 也就是单尾显著度的两倍）——见第三部分的报表。

第四部分的报表将前述的正确检定转以三个检定再检定一次：Z 检定、t（或报表上的

T) 检定、卡平方检定，这三个趋近检定都是双尾的。Z 检定值=3.635=上述的 S 值/标准误。根据常态分配表，此 Z 值的显著度是 0.0003。t 检定的值与 Z 值相等，但其显著度由 t 分配导出，等于 0.0019。第三个趋近检定也就是 Kruskal-Wallis 检定，其显著度由卡平方分配导出，等于 0.0002，自由度=组别数-1=1。这三个趋近检定的结果十分一致，也就是说图形有助幼龄儿童理解故事的内容。

报表 30.2 图形对学生学习的助益

EXAMPL 30.2 DO CHILDREN LEARN BETTER FROM PICTURES OR FROM WORDS 1

N P A R 1 W A Y P R O C E D U R E

Wilcoxon Scores(Rank Sums)for Variable SCORE

Classified by Variable CROUP

第一部分

GROUP	N	Sum of Scores	Expected Under H0	Std Dev Under H0	Mean Score
picture	9	135.0	90.0	12.2420758	15.0000000
verbal	10	55.0	100.0	12.2430758	5.5000000

Average Scores Were Used for Ties

第二部分

Wilcoxon 2-Sample Test S=135.000

第三部分

Exact P-Values

(One-sided) Prob >=s = 1.08E-05

(Two-sided) Prob >=|S-Mean|= 2.17E-05

第四部分

Normal Approximation(with Continuity Correction of .5)

Z=3.63500 Prob>|Z|=0.0003

T-Test Approx.Significance= 0.0019

Kruskal-Wallis Test(Chi-Square Approximation)

CHISQ= 13.512 DF= 1 Prob>CHISQ=0.0002

30.4 注 意 事 项

■ 遗漏数据的处理

若观察体在任何一个 NPAR1WAY 程序的变量上有遗漏数据，则 NPAR1WAY 程序会自动剔除此观察体。

■ 等值数据的处理

等值数据以其所对应的名次平均数为代表。同时，变异数的计算也相对地调整。调整的方式依据 Hajek 的教科书 (1969) 第 7 章的方法为准。